
PREFACE

Intel welcomes you as a new user of the Microcomputer-System/48 (MCS-48) and Universal-Peripheral-
Interface/41 (UP1-41) microcomputer systems. This manual is one of a series of documents describing

these systems and their operation.

Part One of this manual describes the assembly language for programming the families of MCS-48 and
UPI-41 single-device microcomputers. Additional information needed to create a source (input) file to
the 8048/8041 assemblers, specifically the use of assembler directives, is also included in this part of

the manual.

Part Two describes procedures and controls for operating the assemblers used to translate your source

file into object code recognized by the MCS-48 and UPI-41 microcomputer systems. Paper-tape-

resident and diskette-resident versions of the assembler are available on intel’s inteliec Microcomputer

Development System. The Intellec Series |1 Microcomputer Development System provides a ROM-
resident assembler on the Model 210 and a diskette-resident version on the Models 220 and 230.
If you are using the ROM-resident assembler, you will need the document:

Intellec Series |1 Model 210 User’s Guide 9800557

This manual provides only an overview of MCS-48 and UPI-41 hardware and assumes you are familiar

with the documents:
MCS-48 User'’s Manual 9800270
UPI-41 User’s Manual 9800504

If you are not already conversant with the Intellec System and its operation, please refer to the
document:

MDS-800 Intellec Microcomputer Develop-
ment System Operator’s Manual 9800129

If you are using the diskette-resident version of the assembler (ASM48), you will need:
IS1S-11 System User’s Guide 9800306

Finally, you may find the following application notes useful in designing programs for the MCS-48
and UPI-41 microcomputer systems.

Application Techniques for The
MCS-48 Family AP-24

Printer Control Using The UP/-4]1 AP-27

1. FUNCTIONAL OVERVIEW

A microcomputer, until recently, could be defined as a complete computer on a single board. At the heart of the
microcomputer was the microprocessor device, or central processing unit (CPU). The board also contained con-
trol circuitry, memory devices, and input/output (I/O) interfaces.

The MCS-48 and UPI-41 microcomputer systems have made this traditional distinction between microcomputers
and microprocessors obsolete. At the heart of these systems are several single-device microcomputers, each con-
sisting of a CPU, separately addressable program and data memories, |/O interfaces, and timer. The systems are
completed by the addition of applicable Intel peripherals, providing an extensive assortment of family parts. The
MCS-48 microcomputer options are implemented as primary controllers of your OEM equipment. UPI-41 devices
are implemented as intelligent, programmable peripheral processors.

The MCS-48 is available in six functionally similar versions — the 8048 and 8049 microcomputers with read-only
(ROM) program memory, the 8748 microcomputer with erasable and programmable ROM (EPROM), the 8035
and 8039 microcomputers, which use no resident program memory, and the 8021 microcomputer, the lowest
cost component in the MCS-48 family. The UPI-41 is based on either the 8041/8041A microcomputers (with
ROM program memory) or the 8741/8741A microcomputers (with EPROM program memory). The following
chart summarizes the main hardware differences among all eight microcomputers.

External
Microcomputer Pins ROM EPROM RAM Addressing
8048 40 1K — 64 Yes
8748 40 —— 1K 64 Yes
8035 40 — — 64 Yes
8049 40 2K — 128 Yes
8039 40 — — 128 Yes
8022 40 2K — 64 No
8021 28 1K — 64 No
8041A 40 1K — 64 No
8041 40 1K — 64 No
8741 40 — 1K 64 No

These hardware features are discussed in greater detail in the rest of this chapter.

The 8048, 8748, and 8035 are equivalent except for their program memories (ROM/EPROM). The 8035 is used
with external program memories in prototype and preproduction systems. The 8049 and 8039 are also equivalent,
except for program memory, and have the same instruction set as the 8048 group. For the purposes of this
manual, which emphasizes programming primarily, ‘8048’ refers to all five microcomputers.

1-1

Functional Overview

Because of their different usage of the external bus, the 8041, 8741, 8021, and 8022 have a slightly different
instruction set and functional approach from the other five. These differences are discussed at the end of this
chapter and in Chapters 3 and 4. For the purposes of this manual, ‘8041 also refers to the 8741. Descriptions
of the 8048 apply to the 8041, 8741, and 8021 also, except for specifically stated differences.

8048 BASIC FEATURES

From the programmer’s viewpoint, the following are the main 8048 device features:

° Resident 2K or 1K by 8-bit ROM/EPROM program memory with memory
expansion capability

° 128 or 64 by 8-bit random access (RAM) data memory, which includes the
working registers and program counter stack and is also expandable

° 12-bit program counter (PC)

° Program status word (PSW), consisting of status bits, flags, and the stack pointer
° Programmable resident interval timer, also available as an external event counter
° Resident clock and oscillator for internal timing

° External and timer overflow interrupts

° I/O ports and controls, expandable using the 8243 expander device

Program Memory

1-2

Resident program memory consists of a 2K by 8-bit ROM (8049) or a 1K by 8-bit ROM (8048) or EPROM
(8748) divided into 256-byte ‘pages.’ In a typical development sequence, you might program the 8748 with
your prototype code, debug this code using the Intellec system and ICE-48 facilities, and then commit the
final version of your program to the 8048 ROM version for production. Or, you might prefer to use the 8748

for production, leaving yourself the option to make modifications in the field or to tailor your basic program
to customer specifications.

Resident program memory can be expanded up to 4K using additional ROM or EPROM devices. This external
memory is directly addressable by the 8048’s 12-bit program counter. Address selection is done on a ‘bank’
basis using the MCS-48 instructions:

SEL MBO ;SELECT MEMORY BANK O
SEL MB1 ;SSELECT MEMORY BANK 1

Memory bank 0 is the lower 2K of program memory and memory bank 1 is the upper 2K (Figure 1-1). Bits
0-10 of the program counter can address up to 2K locations; PC bit 11 is set to 1 by the SEL MBT instruction,
permitting addressing to 4K. The SEL MB instructions do not affect PC bit 11 until a branch from the main
program sequence is executed (via a call or jump instruction).

. NOTE

Program memory expansion beyond 4K is described in the

MCS-48 user’s manual. Program memory addressing using
the EA (external address) pin is described in the same docu-

ment.
4095
L/\— MEMORY
BANK
-~ \_/_\_
1
2048
. h \\/\J
1024
1023
PAGE 3
768 o MEMORY
767 BANK
PAGE 2 0
512
. 511
PAGE 1
26 | o ____ _
255
PAGE 0O

Figure 1-1 Program Memory Map

’ Data Memory

In addition to resident program memory, the 8049/8039 microcomputers provide a resident 128 by 8-bit data
memory (expandable by 256 locations using additional RAM devices). The other MCS-48 microcomputers have
a 64 by 8-bit resident data memory.

The memory consists of eight working registers (plus an additional eight registers selectable on a ‘bank’ basis), an
eight-level program counter stack, and scratchpad memory (Figure 1-2). The amount of scratchpad memory

available can vary depending on the number of addresses nested in the stack and the number of registers selected.

NOTE

Data memory expansion beyond 256 locations is described in
the MCS-48 user’s manual.

63/127
USER
SCRATCHPAD
MEMORY
/__\
T
<72
31 B R7) REGISTERS 0-7
REGISTER BANK 1 IF BANK 1
24 ] Ro ) 1S SELECTED
23
STACK
8
7 R7
REGISTERS 0—7
REGISTER BANK 0 IF BANK 0
IS SELECTED
0 RO

Figure 1-2 Resident Data Memory Layout

Addressing Data Memory

Working registers in RAM memory can be addressed ‘directly’ by specifying a register number, as in the instruction

MOV A,R4  ;MOVE THE CONTENTS OF REGISTER 4
;INTO THE ARITHMETIC AND LOGIC
UNIT'S 8-BIT ACCUMULATOR

Other locations in resident data memory are addressed ‘indirectly’ using register O or register 1 to specify the
addressed location. The special symbol ‘@’ (commercial at) indicates that indirect addressing is desired.

MOV A,R1T  ;MOVE THE CONTENTS OF REG 1 INTO THE
;ACCUMULATOR

MOV A@R1 ;MOVE THE CONTENTS OF THE LOCATION WHOSE
;ADDRESS IS SPECIFIED BY REG 1 INTO THE
;ACCUMULATOR

Because all 128/64 locations {including the eight working registers) can be addressed by 7/6 bits, the most significant
bits (6 and/or 7) of the addressing registers are ignored. However, all eight bits of register O or register 1 can be used
in combination with the 8048’s MOV X instructions to address up to 256 locations in external RAM data memory.

1-4

MOVX @RO,A ;MOVE THE CONTENTS OF THE ACCUMULATOR
;INTO THAT LOCATION IN EXTERNAL DATA
sMEMORY WHOSE ADDRESS IS CONTAINED
;IN REGISTER O

Working Registers

The dual bank of eight working registers is selected by the 8048’s SEL RB instruction. The initial setting is ‘bank 0,
which refers to data memory locations 0-7. If the instruction

SEL RB1 ;SELECT REGISTER BANK 1

has been issued, then references to RO-R7 in MCS-48 instructions operate on locations 24-31. As was mentioned
above, registers 0 and 1 in the active bank have a special addressing function; they are used to address indirectly all
locations in scratchpad memory (including the optional 256-location expansion). These indirect RAM address
registers are especially useful for repetitive operations on adjacent data memory locations, as in the following
example:

START: ADD A,@R0 ;ADD TO THE ACCUMULATOR THE

‘CONTENTS OF THE LOCATION
‘WHOSE ADDRESS IS SPECIFIED
BY REG 0

INC RO INCREMENT REG 0

JNC START ;JUMP TO INSTRUCTION LABELED
“START’ IF NO ADDITION
OVERFLOW (NO CARRY)

A good programming practice is to reserve locations 24-31 for interrupt servicing routines, thereby preserving the
contents of your main program registers. Simply specify SEL RB1 as one of your interrupt routine’s initialization
instructions. When you subsequently return to the main program using the instruction RETR, the previously

selected bank is automatically restored. During interrupt processing, registers in bank 0 can be accessed indirectly.

Unused registers can serve as additional scratchpad memory, if desired.
Program Counter Stack

Locations 8-23 are used as an 8-level program counter stack. When control is temporarily passed from the main
program to a subroutine or interrupt servicing routine, the 12-bit program counter and bits 4-7 of the program
status word (PSW) are stored in two stack locations (Figure 1-3). Note that the program counter is stored with
its low-order bits in the lowest available address in the stack area.

When control returns to the main program via an RETR instruction, the program counter and PSW bits 4-7 are
restored. Returning via an RET instruction does not restore the PSW bits, however. {These PSW bits are described
in detail later in this chapter.)

The program counter stack is addressed by three stack pointer (STP) bits in the PSW (bits 0-2). The current program
counter is not resident in the program counter stack and consequently is not directly accessible.

7 4 3 0

HIGH (ODD) PS7W PS4W PC11 PCS
LOCATION I | i | [ 1
A ' S

Figure 1-3 Stack Format

The stack pointer bits in the PSW refer to the stack pointer locations as follows:

STP Bits Data Memory Locations

000 8-9

001 10-11
010 12-13
011 14-15
100 16-17
101 18-19
110 20-21
111 22-23

The bit setting indicates the locations to be loaded the next time the program counter is stored. The stack pointer is

incremented by one each time the program counter is stored and decremented cach time the program counter is
restored. Unused stack locations can be employed as scratchpad memory.

The 8048 stack allows up to eight levels of subroutine ‘nesting;’ that is, a subroutine may call a second subroutine,
which may call a third, etc., up to eight levels. When processing interrupts, remember that the stack contains not only
information nested by the main program, but also the program counter stored by the interrupt, plus any information
required by subroutine nesting in the interrupt service routine.

Programmable Controls

The 8048 provides several condition bits, flags, and pins for testing and controlling program operation. These are
referred to as:

C Carry bit

AC Auxiliary carry bit

FO  Flag0 ’
F1  Flag 1

BS  Register bank switch .
TO  Test O pin

T1  Test1pin

TF  Timer flag

i Interrupt input pin

1-6

Carry Bit

The carry bit (C) is affected by the addition and decimal adjust instructions and certain rotate operations and
generally indicates a carry out of the bit 7 position (most significant bit, or MSB) of the ‘accumutator’ (ACC — a
special register in the 8048's arithmetic and logic unit). For example, addition of two 8-bit numbers as in the
following instructions would result in a carry out of the MSB and set the carry bit.

MOV A #0AEH ;MOVE VALUE ‘AE’ HEX TO ACC
ADD A #74H  ;ADD VALUE ‘74’ HEX TO ACC

Bit 7 6 5 4 3 2 7 0

AE 1 0 1 0 1 1 1 0
+74 0 1 1

:1_2.5 [ 0
1 Carry

The carry bit can be complemented (changed to 0 if 1, or to 1 if 0} using the MCS-48 instruction CPL C, reset to
zero using CLR C, and tested by the conditional jump instructions JC and JNC.

o
o
o

o
—_
o
o
ew]
o

Auxiliary Carry Bit

The auxiliary carry (AC) bit indicates a carry out of bit 3 in the accumulator and is only applicable when decimal
arithmetic is being performed. This bit essentially allows the Decimal Adjust Accumulator (DA A) instruction to
perform its function. The DA instruction adjusts the 8-bit accumulator value to form two 4-bit Binary-Coded-

Decimal (BCD) digits. The following instruction sequence resets the carry bit to zero and sets the auxiliary carry
bit.

MOV A #2EH  ;MOVE VALUE “2E’ HEX TO ACC
ADD A,#74H  ;ADD VALUE 74’ HEX TO ACC

Bit 7 6 5 4 3 2 7 0

2E. 0 0 1 0 1 1 1 0
+74 0 1 1

A2 —1 0 1
l—: 0 Carry [ 1 Auxiliary Carry

The auxiliary carry bit cannot be tested or altered directly (but see the discussion of the PSW later in this chapter).
It is affected only by addition.

o
—_
o
o

o
[ew)
o
—_
o

Flag Bits (FO, F1)
The 8048 provides two program control flags (FO and F1), both of which can be complemented with the instructions

CPL FO/F1, reset to zero using CLR FO/F1, or tested with the conditional jumps JFO and JF1. Their initial state is
zero.

1-7

One important difference between these two flags is that FO is restored when control is returned from an interrupt
servicing routine (by the RETR instruction), whereas F1 is not. Therefore the latter can be used by the interrupt
servicing routine to pass an information bit to the main program.

Register Bank Switch

The register bank switch indicates which of the possible register banks (0 or 1) is active. It is toggled by the 8048
instructions SEL RBO and SEL RB1. Its initial state is zero.

Test Input 0

Test input 0 (TO) provides a multifunction capability for the design engineer and programmer. It is directly testable
using the MCS-48 conditional jump instructions JTO and JNTO.

As an input pin activated by an external source it could be used as a pseudo interrupt or other general-purpose
function.

TO can also be converted to a state clock output using the MCS-48 instruction ENTO CLK. This signal could then be
used as a general-purpose clock by the MCS-48. (See the MCS-48 user’s manual for details.)

Test Input 1

A special 8048 register can be used as an interval timer or as an external event counter. As an interval timer it is
initiated by the STRT T instruction and incremented by a prescaler having a periodic duration equivalent to 32
instruction cycles (at 2.5 microseconds per cycle). When the register is used as an event counter, the prescaler is
bypassed and the external test 1 {T1) pin is designated as the counter input. The latter mode is enabled by the STRT
CNT instruction. Both modes are disabled by the STOP TCNT instruction. The conditional jump instructions |T1
and JNTT1 can be used to test this pin.

Timer Flag
As was mentioned above, the interval timer is incremented every 32 instruction cycles. This means the 8-bit timer
register will overflow every 8192 cycles (256 x 32). When the timer overflows, the timer flag (TF) is set, whether
or not the timer overflow interrupt is enabled. The same is true of an event counter overflow (more than 255 T1
inputs).
The timer flag can be tested by the conditional jump instruction JTF. It is reset to zero each time this instruction
is executed. Its initial state is zero.

Interrupt Input Pin

If the ‘external’ interrupt is enabled and this pin is active low (zero level), an interrupt is initiated. (See the dis-
cussion of interrupts below.)

The MCS-48 conditional jump instruction JNI tests for the zero level at this pin. With the interrupt disabled, this
instruction could be used as another test input.

1-8

. Program Status Word

The program status word (PSW) consists of eight bits organized as shown in Figure 1-4,

C {AC | FO | BS 1 |SP2|SP1]|SPO

Figure 1-4 Program Status Word Format

As this figure indicates, locations 4-7 contain the register bank switch (BS), flag 0 (FO), auxiliary carry bit (AC),
and carry bit (C). These four bits are stored in the stack with the program counter when a CALL instruction or an
interrupt is encountered. The bits are restored by an RETR return instruction (but not by RET).

Bits 0-2 of the PSW contain the stack pointer (STP) used to address the 8-level data memory stack (see the sub-
section ‘Program Counter Stack’, above}. Bit 3 of the PSW is unused and is always set to one.

‘ Two MCS-48 instructions (MOV A PSW and MOV PSW, A} allow data to be transferred between the PSW and the
accumulator. This is particularly useful for modifying the stack pointer or AC bits. Bits 4, 5, and 7 can also be
modified individually using the instructions mentioned above (for example, SEL RB1, CLR FO0, CPL C).

Interrupts

The 8048 responds to two kinds of interrupts: ‘external’ and ‘timer overflow.” An external interrupt forces a call to
location 3 in program memory; a timer overflow interrupt forces a call to focation 7.

‘ The external interrupt is enabled by the instruction EN | and disabled by the instruction DIS I. If this interrupt is
enabled and the interrupt input pin goes low (level zero), the interrupt sequence is initiated as soon as the currently

executing instruction is completed. A CALL to location 3 is forced, the return address and bits 4-7 of the PSW are
stored in the program stack, and the stack pointer bits incremented. If you wish, you can create your own ‘interrupt
acknowledge’ by programming an appropriate output pin or by implying the acknowledge in ensuing 1/O operations.

- The RETR instruction should be used to return from an interrupt. This instruction will restore the program counter
and PSW bits 4-7, providing automatic restoration of the previously-active register bank as well. RETR also reenables
| interrupts. On the 8022, the RETI instruction should be used to return from an interrupt.

The timer-overflow interrupt is enabled by the EN TCNTI instruction and disabled by the DIS TCNTI instruction.
if enabled, this interrupt occurs when the timer/event-counter register overflows. A CALL to location 7 is forced
and the interrupt routine proceeds as described above.

After an overflow the timer continues to accumulate time. If you require time intervals greater than the maximum,

‘ you can disable the interrupt, count the number of overflows using the JTF (JUMP if timer flag is one) instruction,
and accumulate the number of overflows in a software counter until the required time is reached. Note that reading
the timer flag with a JTF resets it to zero.



Chapter 1. Functional Overview

While an interrupt service routine is executing, new timer interrupt requests will be accepted, but they cannot be
serviced until the current routine is completed. New external interrupts are not saved. if an external interrupt and a
timer-overflow interrupt occur simultaneously, both are recognized but the external interrupt has highest priority.

NOTE

All routines for servicing interrupts must be located in memory
bank O (program memory locations 0-2K). During servicing of

an interrupt, PC bit 11 is held at zero. The SEL. MB (select memory
bank) instructions should not appear in an interrupt service
routine.

Input/Output
Of the 40 pins on the 8048, 27 can be used for input, output, or both, depending on the MCS-48 configuration

established. In addition to the /O capability provided by these pins, the 8243 expander device can be added to the
configuration to provide 16 additional 1/O lines (four 4-pin ports).

NOTE

I/O expansion beyond that provided by a single 8243 expander
device is described in the MCS-48 user’s manual.

The total 43 1/O lines possible with an 8048 and 8243 expander device are divided into eight directly addressable
groups as follows:

Port Pins Comment

BUS D0-D7 Bidirectional. Strobed
input.

1 P10-P17 Quasi-bidirectional
depending on configuration.

2 P20-P27 P20-23 are used to attach
four 8243 ports. Quasi-

L bidirectional.
TO, T1, INT Testable input pins; test 0,

test 1, interrupt.

4-7 0-15 Four pins each. 8243
ports -

The BUS port and ports 1-2 on the 8048 and ports 4-7 on the 8243 can be read and written by 8048 I/O instructions.
The BUS and ports 1-2 can be ANDed and ORed with the second byte of ANL and ORL instructions. 4

For example:

ANL BUS, #data AND’ SECOND BYTE WITH DATA IN
;BUS PORT

ORL P2,#data ‘OR’ SECOND BYTE WITH DATA IN
PORT 2

Ports 4-7 can be ANDed and ORed with the low-order four bits of the accumulator.

ORLD P5,A ;'OR” ACC BITS 0-3 WITH DATA
;IN PORT 5

Address and control data are provided to the 8243 ports via 8048 pins P20-23. Any data existing on P20-23 before an
8243 instruction is issued is lost. Therefore, if your configuration includes an 8243 expander device, pins P20-23 should
not be used for general 1/O operations.

MICROCOMPUTERS

The 8041 and 8741 (UP1-41) microcomputers are variations of the 8048 and 8748, respectively. The essential
difference between the 8041 and 8048 is that the 8041 includes handshaking interfaces and protocols for MCS-48,
MCS-80, and MCS-85 buses, enabling it to serve as a programmabile, intelligent peripheral within a larger micro-
computing system. This section focuses on the specific design and functional differences between the 8041 and the
8048 required to implement this handshaking. Differences in the assembly language instructions for these devices
are described in Chapter 4.

Functional Differences

During the transfer of data between a master computer and the 8041, the handshaking protocol requires the 8041’s -
BUS port for interfacing to the master port. As a consequence, 8041 program memory cannot be expanded beyond

1K and data memory cannot be expanded beyond 64 locations. 1/O can still be expanded using the 8243 expander
device, however.

The external interrupt function is also committed to the master processor interface. However, the event counter
can provide an effective external interrupt if it is preset to all ones. The T1 input can then be used in the same
manner as the interrupt input, but program control is passed to location 7 rather than location 3 in this instance.

MOV A, #0FFH ;MOVE ‘ONES’ TO ACC

MOV T,A ;MOVE ACC DATA TO TIMER
EN TCNTI ;ENABLE COUNTER INTERRUPT
STRT CNT ;START EVENT COUNTER

In 8041 mode the EN I and DIS | instructions used to enable/disable external interrupts on the 8048 have a
different function. When the master processor is transferring data to the 8041 slave, it can cause an interrupt each
time it fills the 8041’s data bus buffer (described below) to ensure that two writes are not issued before the buffer
is cleared. EN/DIS | enable and disable this interrupt. When initiated, this interrupt passes control to location 3 as
in the normal 8048 external interrupt procedure.

When data is transferred from the 8041 to the master computer, no interrupt is possible except by dedicating /O
lines. The master must poll special 8041 status bits (described below) to determine whether the data bus buffer is

empty.

Finally, the TO pin can be used only as a test input in 8041 mode; it cannot be used as a state clock output.



Chapter 1. Functional Overview

Hardware Differences

Hardware differences (such as pin designation differences, deletion of the functions described above, and hand-
shaking hardware) are described in detail in the UPI-41 user’s manual. However, two special 8041 registers used in
these protocols should be singled out since they are referenced in 8041 instructions.

Data Bus Buffer
The 8-bit data bus buffer (DBB) serves as a temporary register for information flowing between the 8041 and a
master computer. Transfers between the master and slave processors via the data bus buffer can be implemented
with or without program interference (using EN 1 or DIS I).

Data is transferred between the DBB and the 8041’s accumulator using the UPI-41 instructions:

IN A,DBB ;PLACE DBB CONTENTS INTO 8041 ACC
OuUT DBB,A ;PLACE 8041 ACC CONTENTS INTO DBB

Status Register

This 4-bit register indicates the status of flag 0 and flag 1 (FO and F1) and of two special 8041 flags; input buffer
(IBF) and output buffer (OBF). IBF and OBF indicate the condition of the data bus buffer and are initially cleared.

The sequence for transferring data from a master processor to the 8041 is as follows:
° Eight bits are written from the BUS port into the 8041’s DBB .
° IBF is set
° Control/data input is placed in flag 1 (F1)
° An interrupt is generated, if enabled

Subsequent execution of the UPI-47 instruction IN A,DBB in either the main program or the interrupt service

routine clears IBF. The master can determine that IBF has been cleared (that is, DBB is empty and ready for more .

data) by polling the status register. A ‘read control status’ pulse places the 4-bit status register and 4 undefined high-
order bits on the BUS in the order shown in Figure 1-5.

D7 DO

FO IBF | OBF *

T .
| |

Figure 1-5 BUS Contents During Status Polling

When an OUT DBB, A instruction is executed in a UPI-41 program, initiating a transfer of data from the slave to the
master computer, OBF is set. A subsequent ‘read data bus buffer’ pulse from the master reads the DBB contents
onto the BUS and clears OBF.

The slave computer cannot poll or interrupt the master, but it can check the status of the DBB using the two
UPI-41 instructions:

JNIBF addr ;JUMP TO ‘ADDR’ IF IBF NOT SET
JOBF  addr ;JUMP TO ‘ADDR’ IF OBF SET

8041A~8041 MICROCOMPUTER DIFFERENCES

The 8041A UPI microcomputer executes the full instruction set of the 8041 UPI microcomputer, as well as the
following new instructions, described in detail in Chapter 4, under “Added Instructions:”

L] EN DMA (E5 opcode, 1 cycle) enables direct memory access data transfers at high
speeds between the 8041A and the 8257 Programmable DMA Controller.

° EN FLAGS TF5 opcode, 1 cycle) makes OBF and IBF available externally as
service request lines when attached to an interruptable master.

° MOV STS,A (90 opcode, 1 cycle) places accumulator Bits 4 -7 in an 8-bit status
register (STS) for polling by the master. These four bits are user-definable. Bits
0 -3 are unaffected.

Hardware Differences

Details on pin designation differences and handshaking hardware are given in the UP/-471A User’s Manual. Only those
hardware differences relevant to the assembly-language programmer are given here.

Data Bus Buffers

Whereas the 8041 has only one 8-bit data bus buffer (DBB), the 8041A has two, sometimes referred to as “DBBIN”
and “DBBOUT.” However, these informal names are not recognized by the 8041A assembler, which interprets the
intended buffer from the context of the recognized name DBB:

IN A,DBB ;PLACE “DBBIN” CONTENTS IN 8041A ACCUMULATOR
OUTL DBB,A ;PLACE 804TA ACCUMULATOR CONTENTS IN “DBBOUT"

As with the 8041, data transfer can be implemented by polling or interrupt-driven. However, unlike the 8041, the
8041A can interrupt its master as described below.

Status Register

From the viewpoint of the master, the 8041A interface consists of the two data bus buffers just described, and a
third 8-bit status register depicted in Figure 1-5. Bits 4 - 7 of the status register are user-definable, and can be
individually manipulated by the 8041A for use in protocols by executing the instruction:

MOV STS,A ;PLACE 8041A ACCUMULATOR BITS 4 -7 INSTATUS REGISTER
;BITS 0-3 ARE NOT AFFECTED

The master can then interpret these four bits as data, command, or multiprocessing status bits, as defined by the
user.

Interrupting the Master for OBF and IBF

Unlike the 8041, the 8041A can interrupt the master by making OBF (Output Buffer Full) and IBF (Input Buffer
Full — goes low) available externally as service request lines on Ports 24 and 25, respectively. Thus, writinga 1 to
P24 causes OBF to appear and writing a O causes IBF to stay low. These are performed as follows:

EN FLAGS ;ENABLE P24 AND P25 AS EXTERNAL SERVICE REQUEST LINES
ORL P2,#00010000B ;INTERRUPT MASTER TO SIGNAL 8041A OUTPUT BUFFER FULL

The 8041A slave can then poll OBF (as with the 8041) or can be interrupted, if enabled, by the master, which will
have already cleared OBF when it has transferred the output buffer data.

Similarly, the 8041A slave can interrupt the master to inform it that the 804TA input buffer is empty and needs
servicing. This is performed by writing a 1 to P25 (IBF low). If EN FLAGS has already been executed, it does not
need to be repeated. If EN FLAGS has not been executed, the sequence necessary is:

EN FLAGS ;ENABLE P24 AND P25 AS EXTERNAL SERVICE REQUEST LINES
ORL P2,#00700000B ;INTERRUPT MASTER TO SIGNAL 8041A INPUT BUFFER EMPTY

The master can then fill the 8041A input buffer and interrupt the 8041A (IBF will have been set by the master).
If the 8041A is not enabled, it will need to poll IBF in the usual way (using JNIBF as with the 8041).

Performing DMA Requests

Unlike the 8041, the 8041A can request direct memory access (DMA) data transfers by the 8257 by first executing
the instruction:

EN DMA ;ENABLE DMA BY 8257 — HIGH-SPEED DMA CONTROLLER
Once DMA has been enabled, the 804TA can request a DMA cycle by writing a 1 to Port 26:
ORL P2,#01000000B ;WRITE 1 TO P26, BECOMES DRQ

The DMA cycle is then performed by the 8257 and continues until done. Once initiated, the DMA cycle can be
cleared only by RESET.

The 8257 responds at the completion of the DMA cycle by a low-level signal (DACK) to P27. This can be read by
the 8041A using the instruction:

IN A,P2
JB7 LODACK ;JUMP TO SERVICE ROUTINE FOR DMA ACKNOWLEDGE

8021 MICROCOMPUTER

The 8021 is the low-cost, low-end product within the MCS-48 family. It’s features are a subset of the 8048
features described earlier in this chapter. Consequently, a number of instructions in the 8048 instruction set
are not applicable to the 8021 (see Chapter 3 and Appendix A).

Functional Differences

The fundarmental difference between the 8048 and 8021 is in packaging (28 pins on the 8021 vs. 40 pins on
the 8048) and the absence of the BUS port.

The fewer number of pins results in fewer programmable controls and interrupts. The 8021 does contain its
own inboard oscillator, however, and provides the same timer/event-counter capability as the 8048 (using
the T1 test input pin and TF timer flag).

The absence of the BUS port means the 1K on-chip ROM memory and 64-byte RAM memory cannot be
expanded, and 8048 instructions referencing expanded memory are not applicable. Of the 28 pins on the 8021
package, 20 are available for I/O, including /O expansion using the 8243 expander device. The 20 /O lines
possible with an 8021 and 16 expander device lines are divided into the following directly addressable groups:

Port Pins Comment
0 P00-PO7 Quasi-bidirectional with open

drain outputs; optional
pullup device deletion.

1 P10-P17 Quasi-bidirectional.

2 P20-P23 Quasi-bidirectional. Used to
attach four 8243 ports.

- T1 Testable input pin.

4-7 0-15 Four pins each. 8243 ports.

The 8021, like the 8048, provides eight directly-addressable registers (locations 0-7 in RAM memory). All
locations (0-64) in RAM memory can be addressed indirectly through registers 0 and 1. Register bank selection
is not available on the 8021.

Hardware Differences

Hardware differences between the 8048 and 8021 are described in the MCS-48 user’s manual.

8022 MICROCOMPUTER
Like the 8021 microcomputer, the 8022 microcomputer features an 8-bit CPU, on-board clock, ROM, RAM, interval
timer/event counter, zero-cross detection capability, expandable /O, and executes a subset of the 8048 instruction
set.
8022 Microcomputer Hardware Differences
The 8022 microcomputer hardware is considerably enhanced from the 8021:
° ROM is increased from 1024 (1K) bytes on the 8021 to 2K bytes on the 8022.
° The Test 0 {T0) pin is standard on the 8022 for external interrupts.

. On-board analog-to-digital {A/D) conversion is provided on the 8022.

8022 hardware is discussed in detail in the MCS-48 Family of Single-Chip Microcomputers User’s Manual,
Order No. 9800270.

8022 Microcomputer Software Differences

The 8022 executes the full instruction set of the 8021, along with the instructions associated with the following
enhanced capabilities:

Interrupts — ROM locations 3 and 7 are reserved for instructions to be executed upon external
interrupt from TO and T1, respectively, if enabled. If both interrupts occur simultaneously,
the external (TO) interrupt is recognized and control passes to location 3. Unlike the 8048,
the 8022 has no Program Status Word to save. If desired, the accumulator and flags can be
saved using software as described in the example that follows. Note that returns from
interrupt routines should always be performed using the RETI instruction, and not RET.

The interrupt scheme on the 8022 inhibits furhter interrupts beyond the first until RETI

is executed.

Example: 8022 Interrupt Processing — The code that follows shows how accumulator and carry
flags can be saved/restored in your 8022 interrupt routines.

ORG 3 ;PROGRAM MEMORY LOCATION 3 CONTAINS
CALL TOINT ; CALL TO TO INTERRUPT SUBROUTINE
ORG 7 ;PROGRAM MEMORY LOCATION 7 CONTAINS
CALL TTINT ; CALLTO T1T INTERRUPT SUBROUTINE
ORG n ;CODE n TO SUIT APPLICATION
BEGIN: EN/| ;ACTUAL PROGRAM STARTS HERE
EN TCNTI ;BOTH TO AND T1 ENABLED FOR THIS PROGRAM
JMP addr ;DON'T FALL THROUGH TO INTERRUPT ROUTINES
TOINT: MOV R6,A ;TO INTERRUPT — SAVE ACCUMULATOR
CLR A ;PREPARE TO SAVE FLAGS
DA A ;CONVERT CARRY FLAGS TO SIXES
MOV R7,A ;SAVE REPRESENTATION OF FLAGS

;BODY OF TO INTERRUPT ROUTINE

MOV A,R7 ;MOVE REPRESENTATION OF FLAGS TO ACC.

ADD A,#0AAH  ;SET/CLEAR CARRY FLAGS TO PREVIOUS STATE
MOV A,R6 ;RESTORE ACCUMULATOR
RETI ;RETURN FROM INTERRUPT TO NORMAL FLOW
T1INT: MOV R6,A T1 INTERRUPT — SAVE ACCUMULATOR
CLR A ;AND SO ON, AS IN TOINT — ONLY DIFFERENCE IS
; IN BODY

A/D Conversion — Two multiplexed analog inputs are software-selectable using the SEL ANO and

SEL ANT instructions. Only one SEL ANn (n =0,1) can be in effect at a given time, but your program
flow can dynamically switch analog inputs whenever required. Once either analog input has been
selected, A/D conversion reading is performed using the RAD (Read Analog-to-Digital) instruction,
which reads 8 bits of data from the conversion result register into the accumulator.

Example — The following routine alternately reads 8 bits of converted analog data from ANO and
ANT and stores the results in RAM locations 24 -43. Thus, the even locations 24, 26, ..., 42
consist of conversions from ANO, while the odd locations 25, 27, ..., 43 consist of conversions
from ANT. Selection inputs from ANO and AN1 are alternated by manipulating the carry flag.

MOV RO,#24 ;Initialize Reg. O to point to first conversion destination
MOV R1,#20 ;Number of A/D conversions to be read
CLRC ;Initialize carry flag for ANO, ANT selection
READO: SEL ANO ;Select analog input O
READ1: RAD ;Read 8 bits of converted data from selected input
MOV @RO,A ;Store converted data in array starting at 24
INC RO ;Point Reg. O to next array element to receive data
DJNZ RT,MORE  ;Decrement read counter, fall through if done
JMP DONE ;All done
MORE: CPLC ;Complement carry flag for next ANn selection
JNC READO ;Select ANO if no carry
SEL AN1 ;Select ANT if carry flag set
. JMP READI ;Jump to read from ANT using same RAD instruction

DONE:
;Remainder of program

Summary of 8022 Instruction Set Differences (from 8021)
Interrupts and Program Control

EN | — Enable external interrupt

DIS 1 — Disable external interrupt

EN TCNTI — Enable timer/counter interrupt
DIS TCNTI — Disable timer/counter interrupt
RETI — Return from interrupt routine

JTO addr — Jump on TO=1

JNTO addr — Jump on TO=0

Analog-to-Digital Conversion

SEL ANO — Select analog input O
SEL ANT — Select analog input 1
RAD — Move conversion result register to accumulator

Input/Output

IN A,Pp — Input port to accumulator (p=0,1,2)

OUTL Pp,A — Output port to accumulator (p=0,1,2)
MOVD A,Pp — Input 8243 expander port to acc. (p=4-7)
MOVD Pp,A — Output acc. to 8243 expander port (p=4-7)
ANLD Pp,A — AND acc. to 8243 expander port (p=4-7)
ORLD Pp,A — OR acc. to 8243 expander port (p=4-7)

2. ASSEMBLER CONCEPTS

ASSEMBLERS AND ASSEMBLY LANGUAGE

If you have ever written a computer program in a machine-recognizable form such as binary code, you will be par-
ticularly appreciative of the advantages of programming in a symbolic assembly language. Assembly-language
operation codes (opcodes) are easily remembered (for example, MOV for a ‘move’ instruction, JMP for a ‘jump’).
You can also express symbolically the addresses and values referenced in the operand field of assembly language
instructions. The names for these operands can be selected to suggest their purpose, making them as mnemonic as
the opcodes.

The program consisting of assembly language instructions is called a source program. This program is passed through
an assembler, which performs the clerical task of translating symbolic code into object code recognizable by the
MCS-48 and UPI1-41 microcomputers.

The source file passed to the assembler actually includes more than source program instructions. it also includes
assembler directives and (possibly) assembler controls. Only source program instructions are converted into
executable object code, however. The assembler directives and controls initiate various functions that assist and
direct the assembler in its translation operation.

The diskette-resident 8048/8041 assembler, in addition to aliowing symbolic programming, is also a macro
assembler. Frequently repeated routines, identical except for certain parameters, need be coded only once and
thereafter can be generated by a single instruction containing the specific parameters needed. Such routines are
called macros. Macro definition is described in detail in Chapter 6.

Assembler output consists of three possible files: the object file containing your program code in machine-
executable form, the /ist file printout of your source code, object code, and symbol table, and the symbol-cross-
reference file, a listing of symbol-cross-referencerecords. These files are discussed more fully in Part Two.

In this chapter, references to the MCS-48 instruction set apply to the UP1-41 instruction set as well.

INSTRUCTION FORMAT

MCS-48 assembly-language instructions and assembler directives consist of up to four fields as follows:

Labei: Opcode Operand, Operand ;Comment
The label and comment fields are always optional. The operand field may contain zero, one, or two operands
depending on the opcode specified. Any number of blanks can separate fields. The entire instruction must be

entered on one line, terminated by a carriage return and line feed. No continuation lines are possible, though you
may have lines consisting entirely of comments.

2-1

Label Field

An instruction label is a symbol name whose value is the specific memory location where the instruction resides. It

is optional and when present must be followed by a colon. A label can be one to six alphanumeric characters, with
the first character alphabetic. A symbol used as a label cannot be redefined elsewhere in your program. (See ‘Symbols
and Symbol Tables’ later in this chapter.)

Opcode Field

This field contains the mnemonic operation code for the MCS-48 instruction or assembler directive 1o be performed.
It is terminated by a biank or nonalphanumeric character, or by a carriage return and line feed if no operand or
comment field is present.

Operand Field

2.2

The operand field identifies the data to be operated on by the specified instruction opcode. Some instructions re-
quire no operand. Others require one or two operands. In the latter case, the operands are separated by a comma.
As a general rule, when two operands are required (data transfer, addition, and logical operations), the first operand
specifies the destination (or target) of the operation’s result and the second operand specifies the source data.

ADD A,R3 ;ADD CONTENTS OF REG 3 TO ACC

ANL A,R3 ;LOGICAL ‘AND’ CONTENTS OF ACC
;WITH MASK CONTAINED IN REG 3

MOV R1,#0FFH ;MOVE ‘FF’ HEX (ONES) INTO REG 1

Operands can reference directly data contained in MCS-48 registers such as the PSW, accumulator, or data memory
working registers 0-7.

MOV A PSW ;MOVE PSW CONTENTS TO ACC
XCH A,R4 ;EXCHANGE ACC DATA WITH
;REG 4 DATA

All data memory locations can be accessed indirectly by prefacing a reference to Register 0 or T with a ‘commercial
at’ sign (@).

MOV @RO,A ;MOVE ACC CONTENTS TO DATA MEMORY
;LOCATION WHOSE ADDRESS 1S
;SPECIFIED INREG O

The JMPP instruction allows program memory locations to be accessed indirectly by prefacing an accumulator
reference with @.

JMPP @A ;CONTENTS OF PROGRAM MEMORY LOCATION POINTED TO BY
;ACC ARE SUBSTITUTED FOR BITS 0-7 OF PROGRAM COUNTER

Operands can contain ‘immediate’ data. The desired value is inserted directly into the operand field. All immediate
data must be prefixed with a pound sign (#) to distinguish it from register data and must evaluate to eight bits.

Immediate data can be in the form of an ASCII constant (a character enclosed in single quotes), a number, an
expression to be evaluated at assembly time, or a2 symbol name. To indicate a quote as an ASCII constant, show
the quote as two consecutive single quotes (). Any symbol appearing in the operand field must be previously

defined.
MOV A #T’ ;MOVE THE VALUE OF ASCI|I
;CONSTANT “T” (01010100)
;INTO ACC
ADD A,#0AH ;ADD HEX ‘0A” (00001010)
;TO ACC
ANL A, #3+(D/5) ;LOGICAL ‘AND’ CONTENTS OF

;ACCWITH MASK WHOSE VALUE
;IS THE RESULT OF ‘3+(D/5)’

Finally, the operand field of a jump instruction (that is, the address to be jumped to) can be expressed as a sym-
bolic label, as an absolute 12-bit program memory address, or as an expression that can be evaluated to such an
address. In no case is this operand preceded by a pound sign.

JMP START ;JUMP TO THE LOCATION LABELED ‘START’
JMP 200H ;JUMP TO LOCATION 200 HEX (512 DECIMAL)

Expression evaluation and symbols are discussed in more detail in the next two sections of this chapter.

Comment Field

The comment field can contain any information you deem useful for annotating your program. The only stipulation
is that this field be preceded by a semicolon. A double semicolon (;;) preceding a comment in the body of a macro
definition suppresses inclusion of the comment in the macro definition, thus reducing storage requirements.

ARITHMETIC OPERATIONS

When discussing arithmetic operations, we must distinguish between operations performed by your program when
it is executed (such as ADD A,R5) and expression evaluation performed by the assembler at assembly time (such as
MOV A, #P+3%(X/2}. Numbers are represented identically in both cases, but your program has considerably more
flexibility than the assembler in determining the range of numbers, internal notation, and whether numbers are to
be considered signed or unsigned. The characteristics of both modes of arithmetic are summarized in Figure 2-1 and
discussed in more detail in the following subsections. '

2-4

Number Characteristic

Assembly—Time
Expression Evaluation

Program Execution
Arithmetic

Base Representation

Range
Evaluates To:
Internal Notation

Signed/Unsigned
Arithmetic

Binary, Octal, Decimal,
or Hexadecimal

0-(64K-1)
16 Bits
Two’s Complement

Unsigned

Binary, Octal, Decimal,
or Hexadecimal

User Controlled
User Interpretation
Two’s Complement

Unsigned Unless
User Manipulates

Figure 2-1 Number Representation

Number Base Representation

Numbers can be expressed in decimal, hexadecimal, octal, or binary form. A hexadecimal number must begin with a
decimal digit and have the suffix ‘H’ (for example: 3AH, OFFH, 12H). Octal values must have one of the suffixes
‘O’ or ‘Q’ (for example: 760,53Q). Binary numbers must have the suffix ‘B’ {for example: 10111010B). Decimal
numbers can be suffixed optionally by ‘D’ (for example: 512, 512D). Where no suffix is present, decimal is assumed.

Permissible Range of Numbers
In general, numbers can range between 0 and 65,535 (OFFFFH). Numbers outside this range are evaluated ‘modulo’
64K (that is, 2 number greater than 64K is divided by 64K and the remainder substituted for the original number).
All expressions can be evaluated to 16 bits.
Certain limitations must be applied within this general range, however. For example, most program execution arithmetic
is done using the 8-bit accumulator or 8-bit registers and most results evaluate to 8 bits. To work with larger numbers
would require manipulation of register pairs.
If you are doing signed arithmetic, the high-order bit of each number is used to indicate the sign of that number (0 if
positive, 1 if negative). Consequently, the remaining bits can only express a number in the range —32,768 to +32,767
for 16-bit arithmetic. For 8-bit arithmetic, the range is —128 to +127.

If a number is too large for its intended use, either an error results or modulo arithmetic is performed. For example:

° Program memory addresses must be in the range 0-4095 (12 bits). In some cases, an address reference
must be ‘within page,’ that is, within the range 0-255 (8 bits).

° Data memory addresses must be in the range 0-255 (8 bits).




Two’s Complement Arithmetic

° Operands containing 8-bit immediate data must evaluate to an 8-bit number.

° Expressions in a DB assembler directive (except strings) must evaluate to 8 bits.

Two's complement notation allows subtraction to be performed by a series of bit complementations and additions
(thus reducing the circuitry requirements of a processor). A number is converted to two’s complement form by
complementing all its bits and adding a binary one to the result.

When a number is interpreted as a signed two’s complement number, the low-order bits supply the magnitude of
the number and the high-order bit is interpreted as the sign of the number. As was mentioned above, the range of a
signed two’s complement value is —32,768 to +32,767 (for 16 bits) and —128 to +127 (for 8 bits).

When a 16-bit value is interpreted as an unsigned two's complement number, it is considered to be positive and in
the range 0-65,535. An 8-bit value is in the range 0-255.

The assemblers perform all expression evaluation assuming unsigned two’s complement numbers. Similarly, execution-
time arithmetic normally assumes unsigned two’s complement notation, but you can perform signed arithmetic by
isolating and inspecting the high-order bit with the instruction:

JB7 MINUS ;IF ACC BIT 7=1 GO TO ‘MINUS” ROUTINE

The MCS-48 instruction set does not include a subtraction instruction. Subtraction is done by complementing the
accumulator and proceeding as in a normal two’s complement addition operation. The CPL A (complement accumu-
lator) instruction performs a straight binary one’s complement. You must perform the binary addition of one,
necessary to convert the number to two’s complement notation, yourself.

Example: Subtract TAH from 63H using signed two’s complement notation.

MOV A #1AH  ;MOVE ‘TAH’ INTO ACC (00011010)

CPL A ;ONE’S COMPLEMENT ACC (11100101)
INC A ;CONVERT TO TWO’S COMPLEMENT

‘ ;(11100110)
ADD A,#63H  ;ADD ‘63’ TO VALUE IN ACC (01001001)
}B7 MINUS ;IF ACC BIT 7=1 GO TO ‘MINUS’ ROUTINE

The result is +49H.

Assembly—Time Expression Evaluation

An expression is a combination of numbers, symbols, and operators. The latter can be arithmetic, relational, and
logical operators or specially-defined MCS-48 operators. Any symbol appearing in an expression must have a
previously-defined absolute value.

The ASCII characters ‘null” and ‘rubout’ are ignored on input, but the null string can be represented by two consecu-
tive quotes or by a missing operand. The null string is illegal in any context that requires numerical evaluation.

2
Operators

The assembler includes five groups of operators that permit the following assembly-time operations: arithmetic, bit
shifting operations, logical evaluation, value comparison, and byte isolation. These are all assembly-time operations.

Once the assembler has evaluated an expression, it becomes a permanent part of your program.
Arithmetic Operators

The arithmetic operators are as follows:

Operator Meaning
+ Unary or binary addition
— Unary or binary subtraction
* Multiplication
/ Division. Any remainder is discarded (7/3=2)
MOD Modulo. Result is remainder produced by a

division operation (7 MOD 3 = 1)
Examples:
The following expressions generate the bit pattern for the ASCII character A:
5+30%2
(25/5)+30%2
5+ (=30%* -2)
The MOD operator must be separated from its operands by spaces:

NUMBR MOD 8

Assuming that NUMBR has the value 25, this expression evaluates to 1.

Shift Operators

The shift operators are as follows:

Operators Meaning

y SHR x Shift operand ‘y’ to the right ‘x’ bit
positions

y SHL x Shift operand ‘y’ to the left ‘x’ bit
positions

The shift operators do not wrap around any bits shifted out of the byte. Bit positions vacated by the shift operation
are replaced with zeros. The shift operator must be separated from its operands by spaces.

Example:

Assume that NUMBR has the value 0101 0101. The effects of the shift operators is as follows:
NUMBR SHR 2 0001 0101
NUMBR SHL 1 10101010

Shifting one bit position to the left has the effect of doubling a value; a shift one bit position to the right has the
effect of dividing a value in half.

Logical Operators

The logical operators are as follows:

Operator Meaning
NOT Logical one’s compiement
AND Logical AND (=1 if both ANDed bits are 1)
OR Logical OR (=1 if either ORed bit is 1)
XOR Logical EXCLUSIVE OR (=1 if bits are different)

The logical operators act only upon the least significant bit of values involved in the operation. Also, these
operators are commonly used in conditional IF directives. These directives are fully explained in Chapter 5.

Example:
The following IF directive tests the least significant bit of three items. The assembly language code that follows the
IF is assembled only if the condition is TRUE. This means that all three fields must have a one bit in the least

significant bit position.

IF FLD1 AND FLD2 AND FLD3

Compare Operators

The compare operators are as follows:

Operator Meaning
EQ Equal
NE Not equal
LT Less than
LE Less than or equal
GT Greater than -
GE Greater than or equal
NUL Special operator used to test for null

{missing) macro parameters. (IS1S-11
assembler only.)

The compare operators yield a yes-no result. Thus, if the evaluation of the relation is TRUE, the value of the result is
all ones. If FALSE, the value of the result is all zeros. Relational operations are based strictly on magnitude com-

parisons of bit values. Thus, a two's complement negative number (which always has a one in its high order bit) is ‘
greater than a two’s complement positive number {(which always has a zero in its high order bit).

Since the NUL operator applies only to the macro feature, NUL is described in Chapter 6.

The compare operators are commonly used in conditional 1F directives. These directives are fully explained in
Chapter 5.

Notice that the compare operator must be separated from its operands by spaces.

Example:

The following IF directive tests the values of FLD1 and FLD2 for equality. If the result of the comparison is TRUE,
the assembly language coding following the IF directive is assembled. Otherwise, the code is skipped.

IF FLD1 EQ FLD2

Byte Isolation Operators

The byte isolation operators are as follows:

Operator Meaning
HIGH Isolate high-order 8 bits of 16-bit value .
LOW Isolate low-order 8 bits of 16-bit value

As was mentioned in the discussion of number ranges, you will sometimes need an 8-bit address or need to generate an

8-bit value. This is where the HIGH and LOW operators can be useful. .

Examptle:

Assume ADRS is an address manipulated at assembly-time for building tables or lists of items that must all be below
address 255 in memory. The following |F directive determines whether the high-order byte of ADRS is zero, indi-
cating the address is still less than 256:

IF HIGH ADRS EQ 0

Precedence of Operators

Expressions are evaluated left to right. Operators with higher precedence are evaluated before other operators that
immediately precede or follow them. When two operators have equal precedence, the leftmost is evaluated first.

Parentheses can be used to override normal rules of precedence. The part of an expression enclosed in parentheses is
evaluated first. If parentheses are nested, the innermost are evaluated first.

15/3+18/9 5+2=7

15/(3 +18/9) 15/(3+2)=15/5=3

The following list describes the classes of operators in order of precedence:
® Parenthesized expressions

NUL

HIGH, LOW

Multiptication/Division: *, /, MOD, SHL, SHR

Addition/Subtraction: +, — (Unary and binary)

Relational Operators: EQ, LT, LE, GT, GE, NE

Logical NOT

Logical AND

Logical OR, XOR

The relational, logical, and HIGH/LOW operators must be separated from their operands by at least one blank.

SYMBOLS AND SYMBOL TABLES
Symbolic Addressing

If you have never done symbolic programming before, the following analogy may help clarify the distinction between
a ‘symbolic’ and an ‘absolute’ address.

The locations in program memory can be compared to a cluster of post office boxes. Suppose Richard Roe rents box

500 for two months. He can then ask for his letters by saying ‘Give me the mail in box 500,” or ‘Give me the mail for
Roe.’ If Donald Doe later rents box 500, he too can ask for his mail by either box number 500 or by his name.

29

The content of the post office box can be accessed by a fixed, abso/ute address (500) or by a symbolic, variable
name. The postal clerk correlates the symbolic names and their absolute values in his log book. The MCS5-48 clerk,
the assembler, performs the same function, keeping track of symbols and their values in a symbo/ table. Note that
you do not have to assign values to symbolic addresses. The assembler references its location counter during the
assembly process to calculate these addresses for you. (The location counter does for the assembler what the pro-
gram counter does for the microcomputer. It tells the assembler where the next instruction or operand is to be
placed in memory.)

Symbol Characteristics
A symbol can contain one to six alphabetic (A-Z) or numeric (0-9) characters (with the first character alphabetic)

or the special character ‘7. A dollar sign can be used as a symbol to denote the value currently in the location
counter. For example, the command

JMP $+6 ‘

forces a jump to the instruction residing six memory locations higher than the JMP instruction. Symbols of the form
‘2?7nnnn’ are generated by the assembler to uniguely name symbols local to macros.

The assemblers regard symbols as being reserved or user-defined, global or limited, permanent or redefinable. All
MCS-48 symbols are absolute, that is, fixed to some absolute memory address or fixed-value expression unaffected

by program loading. .

Reserved, User-Defined, and Assembler-Generated Symbols

The ‘$’ symbol and following MCS-48 and UPI-41 instruction-set opcodes are reserved and cannot be specified as
user-defined symbols except in a limited context (as macro dummy parameters or as symbols defined as local to a
macro definition).

ADD ENTO JNI MOVD RL

ADDC IN JNIBF MOVP RLC

ANL INC JNTO MOVP3 RR

ANLD INS JNT1 MOVX RRC

CALL JBn JNZ NOP SEL

CLR 1 JOBF ORL STOP

CPL JFO JTF ORLD STRT

DA JF1 JTO ouT SWAP .
DEC JMP IT1 OuTL XCH

DIS JMPP Y4 RET XCHD

DINZ JNC MOV RETR XRL »
EN

The following instruction operand symbols and symbols required by the assembler are also reserved:

RO
R1
R2
R3
R4
RS
R6
R7
PSW
BUS
PO
P1
P2
P4
P5
P6
P7

CNT
TCNT
RBO
RB1
MBO
MB1

TCNTI
FO

F1
DBB
ANO
ANT1
FLAGS
RAD
STS

Finally, the following directives cannot be used as symbols except in a limited context:

DB END
DS ENDIF
DW ENDM
ELSE EOT

Meaning

Accumulator

Register O

Register 1

Register 2

Register 3

Register 4

Register 5

Register 6

Register 7

Program Status Word
BUS Port

1/O Port 0 (8021)

1/O Port 1

1/O Port 2

1/O Port 4

1/O Port 5

1/O Port 6

i/O Port 7

Carry Flag

Timer Register

Counter Register
Timer/Counter

Register Bank 0
Register Bank 1
Memory Bank O
Memory Bank 1
Interrupt
Timer/Counter Interrupt
Flag O

Flag 1

Data Bus Buffer (8041)
Analog Input 1 (8022)
Analog Input 2 (8022)
Service Request Lines (8041A)
Read Analog-to-Digital (8022)
Status Register (8041A)

EQU IRPC ORG
EXITM LOCAL  REPT

MACRO  SET

IRP

Chapter 2. Assembler Concepts

User-defined symbols are symbols you create to reference instruction addresses and data. These symbols are defined
when they appear in the label field of an instruction or in the name field of EQU, SET, or MACRO assembler
directives (see Chapters 5 and 6). Values for these symbols are determined modulo 64K although specific environ-
ments may limit the value even further. (See the subsection ‘Permissible Range of Numbers,’ earlier in this chapter.)

Values outside these ranges cause an error.

Assembler-generated symbols are created by the assembler to replace user-defined symbols that have limited scope
(limited to a macro definition).

NOTE

Only instructions that allow registers as operands may have

register-type operands. Expressions containing register-type

operands are flagged as errors. The only assembler directives

that may contain register-type operands are EQU, SET, and

actual parameters in macro calls. Registers can be assigned s
alternate names only by EQU or SET.

Global and Limited Symbols

Symbols appearing as dummy parameters in a macro definition have limited scope and may only be used within
that macro definition. Other symbols appearing in the body of a macro definition can be specified to have limited
scope using the LOCAL assembler directive.

All other symbols, including macro definition names, have global scope and can be referenced from any part of your
program. However, nested macro names cannot be called until all higher-level nested definitions have been called.

Permanent and Redefinable Symbols

Most symbols are permanent, that is, their values cannot change during the assembly operation. Only symbols
defined with the SET and MACRO assembler directives are redefinable.

Duplicate Symbols

Local symboil names can be the same as reserved symbols, or local symbol names in other macro definitions. The
assembler assigns a unique name to each local symbol.

A macro body containing a global label can be called only once. Additional calls cause ‘multiply-defined symbol’
errors. Attempts to redefine local or global symbols (other than with the SET directive) cause the same error.

3. MCS -48 ASSEMBLY LANGUAGE INSTRUCTIONS

This chapter describes the instruction set for the 8048, 8748, 8035, 8049, 8039, 8022, and 8021 (MCS-48)
microcomputers. The 8041, 8041A, 8741, and 8741A (UPI-41) microcomputers use essentially the same
instructions. The few differences are described in Chapter 4.

The instructions are described here in four main functional groupings:

° Data Transfer
— Within memory
— Input/Qutput
° Data Manipulation

- Logical operations

— Bit rotation (shift)

— Arithmetic

-- Miscellaneous accumulator operations

° Setting Program Controls
— Condition bits, flags
- Timer/event counter
— Interrupts
-- Register and memory banks
— NOP

° Transferring Program Control
- Subroutine call
— Return from subroutine
-- Jump operations

Most MCS-48 instructions require one machine cycle for execution (2.5 microseconds for the 8048, 10 micro-
seconds for the 8021 and 8022). Exceptions are I{/O instructions, instructions using immediate data, subroutine
calls and returns, jumps, and certain data transfers within memory, which require two cycles.

NOTE

For microcomputers having more than 1K of program
memory, the only instructions that can reside in the
last byte of a 2K block {locations 2047, 4095) are the
subroutine returns {(RET, RETR) and the second byte
of a jump instruction. Exceptions cause a displacement
(D) error. See Appendix F.

DATA TRANSFER INSTRUCTIONS
Data Transfer Within 8048 Memory

This section describes those instructions used to move data within resident and external 8048 data memory and pro-
gram memory. This includes the MOV, MOV X, and MOVP data move instructions and the XCH and SWAP data
exchange instructions. The move instructions overlay existing data in the target location. Data in the source location
is unchanged. The exchange instructions swap data between two locations.

Register/Accumulator Moves

Data can be transferred between 8048 data memory working registers 0-7 and the accumulator by addressing the
registers directly (R0-R7). R0O-R7 can refer to data memory locations 0-7 if register bank 0 has been selected or to
locations 24-31 if register bank 1 has been selected. Register bank 0 is the initialization value.

Move Register Contents to Accumulator

Op Code Operands

MOV A Rr r=0-7

T1T1T 1M Vrrr

Eight bits of data are moved from working register ‘r’ into the accumulator.
Example:

MAR: MOV AR3 ;MOVE CONTENTS OF REG
;3 TO ACC

Move Accumulator Contents to Register

Op Code Operands

MOV Rr,A r=0-7

101041 rrr

The contents of the accumulator are moved to register ‘r’. o
Example:

MRA: MOV RO,A  ;MOVE CONTENTS OF
SACCTOREG O

Data-Memory[Accumulator Moves

Data moves between the accumulator and nonregister locations in data memory are accomplished by placing the
address of the memory location in either register O or register 1 of the currently selected register bank. This is
called indirect addressing. The assembler knows that indirect addressing is intended by the ‘commercial at’ sign
(@) preceding the register reference.

The MOV instructions reference locations 0-63 (8048) or locations 0-127 (8049) in resident data memory. The
MOV X instructions reference locations 0-255 in the optional external data memory.

Move Data Memory Contents to Accumulator
Opcode Operands

MOV A,@Rr r=0-1

11T11000r

The contents of the data memory location addressed by bits 0-5 (8048) or bits 0-6 (8049) of register ‘r’ are
moved to the accumulator. Register ‘r’ contents are unaffected.

Example: Assume R1 contains 00110110.
MADM: MOV A @RI iMOVE CONTENTS OF DATA MEM
;LOCATION 54 TO ACC
Move Accumulator Contents to Data Memory
Opcode Operands

MOV @Rr,A r=0-1

1070000

The contents of the accumulator are moved to the data memory location whose address is specified by bits 0-5
(8048) or bits 0-6 (8049) of register ‘r’. Register ‘r’ contents are unaffected.

Example: Assume RO contains 00000111,

MDMA: MOV  @RO,A ;MOVE CONTENTS OF ACC
;TO LOCATION 7 (REG 7)

34

Move External-Data-Memory Contents to Accumulator
Opcode Operands

MOVX A,@Rr r=0-1

1000000

This is a 2-cycle instruction. The contents of the external data memory location addressed by register ‘r’ are

moved to the accumulator. Register ‘r’ contents are unaffected. This instruction is not recognized by the
8021 or 8022.

Example: Assume R1 contains 01110110.

MAXDM: MOVX A@R1 ;MOVE CONTENTS OF
;LOCATION 118 TO ACC

Move Accumulator Contents to External Data Memory
Opcode Operands

MOVX @Rr,A r=0-1

1001(000r

This is a 2-cycle instruction. The contents of the accumulator are moved to the external data memory location
addressed by ‘r’. Register ‘r’ contents are unaffected. This instruction is not recognized by the 8021 or 8022.

Example: Assume RO contains 11000111,

MXDMA: MOVX @RO,A ;MOVE CONTENTS OF ACCTO
;LOCATION 199 IN EXTERNAL
;DATA MEMORY

Immediate-Data Moves

Data can be inserted directly into the accumulator, a working register, or resident data memory using the move-
immediate-data instructions. Immediate data can be in the form of an ASCII constant, a number, an expression to
be evaluated at assembly time, a symbol name, or an instruction enclosed in parentheses. (See Chapter 2, the sub-
section ‘Operand Field.’) The assembler recognizes immediate data by the ‘pound sign’ (#) preceding such data.

Immediate data must evaluate to a number that can be expressed in eight bits (that is, less than 256 decimal).

Larger numbers generate an error condition. Larger numbers can be placed in data memory, however, by moving
immediate data to adjoining locations.

Move Immediate Data to Register

Opcode Operands
MOV Rr,#data r=0-7
101 11T rrr data

This is a 2-cycle instruction. The 8-bit value specified by ‘data’ is moved to register ‘r.’
Examples:

MIR4: MOV  R4,#EXTEN ;THE VALUE OF THE SYMBOL
S HEXTEN’ IS MOVED INTO
;REG 4

MIR5: MOV  R5#PI*(R*R) ;THE VALUE OF THE
;EXPRESSION ‘PI*(R*R) IS
;MOVED INTO REG 5

MIR6: MOV  R6,#0ACH ;AC HEX IS MOVED INTO
;REG 6

Move Immediate Data to Data Memory
Opcode Operands

MOV @Rr #data r=0-1

10111000 data

This is a 2-cycle instruction. The 8-bit value specified by ‘data’ is moved to the resident data memory location
addressed by register ‘r,’ bits 0-5 (8048) or bits 0-6 (8049).

Example: Move the hexadecimal value AC3F to locations 62-63.

MIDM: MOV  RO,#62 ;MOVE ‘62’ DEC TO ADDR REG 0
MOV  @RO,#0ACH ;MOVE ‘AC’ HEX TO LOCATION 62
INC RO ;INCREMENT REG 0 TO ‘63’
MOV  @RO,#3FH ;MOVE 3F’ HEX TO LOCATION 63

Move Immediate Data to Accumulator

Opcode Operands
MOV A #data
0010/0011 data

This is a 2-cycle instruction. The 8-bit value specified by ‘data’ is moved to the accumulator.

Example:
MOV A #0A3H ;MOVE ‘A3’ HEX TO ACC
-
PSW/Accumulator Moves
Data can be moved back and forth between the program status word and the accumulator. This is particularly use- -

ful for manipulating the stack pointer (PSW bits 0-2), which cannot be altered by specific instruction (as can the
carry, flag 0, and register bank switch bits in the PSW}.

Move PSW Contents to Accumulator
Opcode Operands

MOV APSW

110010111

The contents of the program status word are moved to the accumulator. This instruction is not recognized by
the 8021 or 8022.

Example: Jump to ‘RBT1SET’ routine if PSW bank switch, bit 4, is set.

BSCHK: MOV  A,PSW ;MOVE PSW CONTENTS TO ACC
|B4 RBTSET ;JUMP TO ‘RBTSET’ IF ACC
;BIT 4=1

Move Accumulator Contents to PSW

Opcode Operands

MOV PSW,A

110170111

The contents of the accumulator are moved into the program status word. All condition bits and the stack pointer
are affected by this move. This instruction is not recognized by the 8021 or 8022.

&
Example: Move up stack pointer by two memory locations, that is, increment the pointer by one.
INCPTR: MOV  APSW ;MOVE PSW CONTENTS TO ACC
INC A JINCREMENT ACC BY ONE
MOV  PSW A ;MOVE ACC CONTENTS TO PSW

Timer[Accumulator Moves

Data can be moved between the accumulator and the special timer/event-counter register. This allows initialization
and monitoring of this register’s contents.

Move Timer/Counter Contents to Accumulator

Opcode Operands

MOV AT

01 00/0010

The contents of the timer/event-counter register are moved to the accumulator.
Example: Jump to ‘EXIT’ routine when timer reaches ‘64,’ that is, when bit 6 is set — assuming initialization (64.
TIMCHK: MOV AT ;MOVE TIMER CONTENTS TO ACC
JB6 EXIT ;JUMP TO ‘EXIT’ IF ACC BIT 6=1
Move Accumulator Contents to Timer/Counter
Opcode Operands

MOV TA

The contents of the accumulator are moved to the timer/event-counter register,

Example: Initialize and start event counter.

INITEC: CLR A ;CLEAR ACC TO ZEROS
MOV  T,A ;MOVE ZEROS TO EVENT COUNTER
STRT CNT ;START COUNTER

Program-Memory [Accumulator Moves

Data in program memory can be accessed indirectly using the accumulator as an address register. The accumulator
reference is preceded by a ‘commercial at’ sign (@) to indicate indirection. The 8-bit address in the accumulator is
used to reference a location in program memory; the contents of the memory location are then moved to the
accumulator. These instructions are not recognized by the 8021 or 8022.

The 8-bit address limits the range of a program memory reference to the current 256-location page. One special

instruction allows you to reference page 3 (locations 768-1023) from any location in program memory, however.
This convenience lets you group frequently-accessed information {such as tables or indexes) in one easily-read area.

Move Current Page Data to Accumulator
Opcode Operands

MOVP ACA

10100071 1

The contents of the program memory location addressed by the accumulator are moved to the accumulator. Only
bits 0-7 of the program counter are affected, limiting the program memory reference to the current page. The pro-
gram counter is restored following this operation. This instruction is not recognized by the 8021 or 8022.

NOTE

This isa 1-byte, 2-cycle instruction. If it appears in
location 255 of a program memory page, @A addresses
a loeation in the foflowing page.

Example:

MOV128: MOV A#128  ;MOVE ‘128’ DECTO ACC
MOVP A,@A ;CONTENTS OF 129TH LOCATION
;INCURRENT PAGE ARE MOVED

;TO ACC
Move Page 3 Data to Accumulator
Opcode Operands
MOVP3 A,@A

T 110000 11

This is a 2-cycle instruction. The contents of the program memory location (within page 3) addressed by the
accumulator are moved to the accumulator. The program counter is restored following this operation. This
instruction is not recognized by the 8021 or 8022.

Example: Look up ASCII equivalent of hexadecimal code in table contained at the beginning of page 3. Note that
ASCII characters are designated by a 7-bit code; the eighth bit is always reset (see Appendix E).

TABSCH: MOV A #0B8H ;MOVE ‘B8’ HEX TO ACC (107111000)

ANL A#IFH  ;LOGICAL AND ACC TO MASK BIT 7
;(00111000)
MOVP3  A@A ;MOVE CONTENTS OF LOCATION ‘38’

;HEX IN PAGE 3 TO ACC (ASCII ‘8")

Example: Access contents of location in page 3 labeled TABT1. Assume current program tocation is not in page 3.
NOTE: The LOW operator is described in Chapter 2, ‘Assembly-Time Expression Evaluation.’

TABSCH: MOV A#LOW TAB1 ;ISOLATE BITS 0-7 OF LABEL
;ADDRESS VALUE
MOVP3  A,@A ;MOVE CONTENTS OF PAGE 3 LOCATION

;LABELED ‘TAB1 TO ACC

Data Exchange Instructions
Data can be exchanged between the accumulator and working registers specified directly, or between the
accumulator and data memory locations specified indirectly (preceded by @). The exchange instructions apply
only to resident data memory, and not to external memory.
The main advantage of a data exchange over a simple move is that data at the target location is not lost and can
be moved back to its original location if necessary. Binary Coded Decimal {BCD) arithmetic can be performed on
the 8048 by dividing 8-bit values into two 4-bit BCD digits. Two instructions, XCHD and SWAP, allow transfer of
such 4-bit digits.
Exchange Accumulator-Register Contents

Opcode Operands

XCH A,Rr r=0-7

The contents of the accumulator and the contents of working register ‘r’ are exchanged.

Example:

Move PSW contents to Reg 7 without losing accumulator contents.

XCHART7: XCH  AR7 ;EXCHANGE CONTENTS OF REG 7
;AND ACC
MOV  APSW ;MOVE PSW CONTENTS TO ACC
XCH  A,R7 ;EXCHANGE CONTENTS OF REG 7

;AND ACC AGAIN

Exchange Accumulator and Data Memory Contents
Opcode Operands

XCH A,@Rr r=0-1

001 0(00O0T

The contents of the accumulator and the contents of the data memory location addressed by bits 0-5 (8048) or bits
0-6 (8049) of register ‘r’ are exchanged. Register ‘r’ contents are unaffected.

Example: Decrement contents of location 52.

DEC52: MOV  RO,#52 ;MOVE ‘52’ DEC TO ADDRESS
;REG O
XCH  A,@RO ;EXCHANGE CONTENTS OF ACC
;AND LOCATION 52
DEC A ;DECREMENT ACC CONTENTS
XCH  A,@RO ;EXCHANGE CONTENTS OF ACC

;AND LOCATION 52 AGAIN

Exchange Accumulator and Data Memory 4-Bit Data

Opcode Operands

XCHD A @Rr r=0-1

0O 01T 10 0 0 r

This instruction exchanges bits 0-3 of the accumulator with bits 0-3 of the data memory location addressed by bits
0-5 (8048) or bits 0-6 (8049) of register ‘r.” Bits 4-7 of the accumulator, bits 4-7 of the data memory location, and
the contents of register ‘r’ are unaffected.

Example: Assume program counter contents have been stacked in locations 22-23.

XCHNIB: MOV RO,#23  ;MOVE 23’ DECTO REG 0
CLR A ;CLEAR ACC TO ZEROS
XCHD A,@RO ;EXCHANGE BITS 0-3 OF ACC
;AND LOCATION 23 (BITS
;8-11 OF PC ARE ZEROED,
;ADDRESS REFERS TO PAGE 0)

Swap 4-Bit Data Within Accumulator
Opcode Operand

SWAP A

Bits 0-3 of the accumulator are swapped with bits 4-7 of the accumulator.

PCKDIG: MOV
MOV
XCHD

> SWAP

XCHD

MOV

. Input/OQutput Data Transfers

RO,#50
R1,#51
A,@RO
A
A@RT

@RO,A

Chapter 3. MCS-48 Assembly Language Instructions

Example: Pack bits 0-3 of locations 50-51 into location 50.

;MOVE ‘S0’ DECTO REG 0
;MOVE ‘51" DEC TO REG 1
;EXCHANGE BITS 0-3 OF ACC
;AND LOCATION 50

;SWAP BITS 0-3 AND 4-7 OF
JACC

;EXCHANGE BITS 0-3 OF ACC
;AND LOCATION 51

;MOVE CONTENTS OF ACCTO
;LOCATION 50

The MCS-48 input/output instructions allow data to be transferred between the accumulator and /O ports. As was
described in Chapter 1 (the subsection ‘Input/Output’), the BUS port and ports 0-2 are used for standard |/O
operations. Ports 4-7 on the 8243 expander, consisting of four pins each, can be attached through port 2, pins
P20-23, to provide 16 additional 1/O lines. Port O is used only by the 8021, as it does not have a BUS port.

] The 8022 uses ports 0, 1, and 2 as 8-line input, output, or quasi—bidirectional ports.

. All input/output data transfers are 2-cycle operations.

Standard 1/O Transfers

The BUS port and ports 0-2 can be either input or output ports depending on the instruction flow. The BUS

port actually has two modes of operation. If the MCS-48 is used as a freestanding device, the BUS acts as a general
1/O port like ports 0-2. If the MCS-48 is part of a more extensive system with expanded memory and 1/O, the BUS

is a bidirectional port with synchronous strobes. Bus lines are latched only for single-device (freestanding) operations.

Input Port 0-2 Data to Accumulator

Opcode Operands
IN A Pp p=0-2
0O 0 0 0|1 O p p

& ‘ Data present on port ‘p’ is transferred (read) to the accumulator. Only port O is available on the 8021. Ports

0-2 are available on the 8022.

Example:
INP12: IN A,P1
‘ MOV  R6,A
IN AP2
MOV  R7,A

JINPUT PORT 1 CONTENTS TO ACC
;MOVE ACC CONTENTS TO REG 6
;INPUT PORT 2 CONTENTS TO ACC
;MOVE ACC CONTENTS TO REG 7

OFFH (ones) shouid be written to ports 1 and 2 before using them as inputs (using the OUTL Pp,A instruction des-
cribed below.

Strobed Input of BUS Data to Accumulator
Opcode Operands

INS A,BUS

0 00 0f1 0 0O

Data present on the BUS port is transferred (read) to the accumulator when the RD pulse is dropped. (Refer to
timing diagrams in the MCS-48 user’s manual for details.) This instruction is not recognized by the 8027 or 8022.

Example:

INPBUS: INS A,BUS ;INPUT BUS CONTENTS TO ACC

Output Accumulator Data to Port 0-2

Opcode Operands Opcode Operands
OUTL PO, A OouTL Pp, A p=1-2
1 0 0 110 0 0 O 001 1{1 0 p p

Data residing in the accumulator is transferred (written) to port ‘p’ and latched. Port ‘0’ can be specified only
for the 8021. This instruction is not recognized by the 8022.

Example:
OUTLP: MOV A,R7 ;MOVE REG 7 CONTENTS TO ACC
OouTL P2,A ;OUTPUT ACC CONTENTS TO PORT 2
MOV AR6 ;MOVE REG 6 CONTENTS TO ACC
OUTL P1,A ;OUTPUT ACC CONTENTS TO PORT i

Output Accumulator Data to BUS
Opcode Operands

OUTL BUS,A L

0 00 0J0 0 1 O

Data residing in the accumulator is transferred {written) to the BUS port and latched. The latched data remains valid
until altered by another OUTL instruction. Any other instruction requiring use of the BUS port (except INS) des-
troys the contents of the BUS latch. This includes expanded memory operation (such as the MOV X instruction).

Logical operations on BUS data (AND and OR} assume the QUTL BUS A instruction has been issued previously.
This instruction is not recognized by the 8021 or 8022.
Example:

OUTLBP: OUTL BUSA ;OUTPUT ACC CONTENTS TO BUS

Output Accumulator to Port 0-2 (8021 and 8022 only)

This instruction latches 8 bits of data from the accumulator to the specified port. This is a 1-byte instruction
requiring 2 cycles. The instruction takes 3 forms, as follows:

Instruction Opcode Meaning

OuUTL PO,A 90H Latch accumulator data to Port O
OUTL P1,A 39H Latch accumulator data to Port 1
OUTL P2,A 3AH Latch accumulator data to Port 2

Analog-to-Digital Conversion (8022 only)

Three 8022 instructions provide access to the 8022 on-chip A/D conversion feature, described with an example in
Chapter 1 under the heading <8022 Microcomputer.” Each instruction is a single byte.

Instruction Opcode Cycles Meaning

SEL ANO 85H 1 Select analog input O

SEL ANI1 95H 1 Select analog input 1

RAD 80H 2 Read Selected conversion to

accumulator

These instructions are not recognized by the 8021/8039/8049/8048 or 8041/8041A.

Expanded 1/0O Transfers

Data can be transferred between the accumulator and ports 4-7 on the 8243 expander device using the MOVD
instructions. The 8243 attaches to port 2 pins P20-23 and existing P20-23 data is destroyed by these instructions.

Ports 4-7 are four pins each. The MOVD instructions transfer data to/from bits 0-3 of the accumulator.

Move Port 4-7 Data to Accumulator
Opcode Operand’s

MOVD APp p=4-7

0O 0 0 O0O|1T 1 pp

Data on 8243 port ‘p’ is moved (read) to accumulator bits 0-3. Accumulator bits 4-7 are zeroed.

NOTE

Bits O0-1 of the opcode are used to represent ports
4-7. If you are coding in binary rather than assem-
bly language, the mapping is as foilows:

Bits1 0 Port
00 4 »
01 5
10 6
11 7 -
Example:
INPPT5:  MOVD A,P5 ;MOVE PORT 5 DATA TO ACC
;BITS 0-3, ZERO ACC BITS
47
Move Accumulator Data to Port 4, 5, 6, or 7
Opcode Operand's
MOVD Pp,A p=4-7

0 01T 111 1 p p

Data in accumulator bits 0-3 is moved {written) to 8243 port ‘p.” Accumulator bits 4-7 are unaffected. (See NOTE
above regarding port mapping.)

Example: Move data in accumulator to ports 4 and 5.

OUTP45: MOVD P4,A ;MOVE ACC BITS 0-3 TO PORT 4
SWAP A ;EXCHANGE ACC BITS 0-3 AND

A4-7
MOVD P5,A ;MOVE ACC BITS 0-3 TO PORT 5

DATA MANIPULATION INSTRUCTIONS

The MCS-48 instruction set includes 34 instructions for manipulating data including logical operations, bit rotation,
incrementing and decrementing of data, addition, and miscellaneous accumulator operations.

Logical Operations

Operations in this category include logical AND, OR, and EXCLLUSIVE OR {XOR). Assuming an initial value of
11100111, a mask of 10101010 would produce the following results following these operations.

11100111 11100111 11100111
AND 10101010 OR 10101010 XOR 10101010
10100010 11101111 01001101
(=1 if both (=1 if either (=1 if bits
are 1) is 1) different)

Most of the logical instructions operate on values in the accumulator. However the 8048 aiso allows logical AND and
OR operations on data residing in 1/O ports.

Accumulator/Register Logical Operations

In the following three instructions, the specified working register contains the mask to be combined logically with an
accumulator value. The result of the operation remains in the accumulator.

Logical AND Accumulator With Register Mask

Opcode Operands
ANL A,Rr r=0-7

O 1 o0 111 r r r

Data in the accumulator is logically ANDed with the mask contained in working register ‘r.’

Example:

ANDREG: ANL  A,R3 AND” ACC CONTENTS WITH
;MASK IN REG 3

Logical OR Accumulator with Register Mask

Opcode Operands
ORL ARr r=0-7

0 1 0 O|1T r r r

Data in the accumulator is logically ORed with the mask contained in working register ‘r.’
Example:

ORREG: ORL A,R4 ;OR” ACC CONTENTS WITH
;MASK IN REG 4

Logical XOR Accumulator With Register Mask

Opcode Operands
XRL A,Rr r=0-7

T 1T 0 1|1 r r r

Data in the accumulator is EXCLUSIVE ORed with the mask contained in working register ‘r.’

Example:

XORREG: XRL AR5 ;' XOR” ACC CONTENTS WITH MASK IN
;REG 5

Accumulator/Data-Memory Logical Operations .
The mask for a logical operation can reside anywhere in resident data memory. (Logical operations cannot reference
external memory.) The address of the mask source is contained in Register O or Register 1. Indirect addressing is

indicated by the ‘@’ preceding the register reference.

The value to be masked and result reside in the accumulator.

Logical AND Accumulator With Memory Mask

Opcode Operands
ANL A,@Rr r=0-1

01 0 10 0 0 r

Data in the accumulator is logically ANDed with the mask contained in the data memory location referenced by
register ‘r,’ bits 0-5 (8048) or bits 0-6 (8049).

Example:

ANDDM: MOV  RO#0FFH ;MOVE ‘FF’ HEX TO REG 0
ANL A @RO FAND’ ACC CONTENTS WITH MASK
;IN LOCATION 63

Logical OR Accumulator With Memory Mask

Opcode Operands

ORL A,@Rr r=0-1

01 0 010 0 O r

Data in the accumulator is logically ORed with the mask contained in the data memory location referenced by
register ‘r,” bits 0-5 (8048) or bits 0-6 (8049).

Example:
ORDM: MOV  RO,#3FH iMOVE 3F'HEX TO REG 0
ORL  A,@RO ;'OR’ ACC CONTENTS WITH MASK
JIN LOCATION 63 )
Logical XOR Accumulator With Memory Mask &
Opcode Operands
XRL A,@Rr r=0-1

t 1.0 1{0 0 0 r

Data in the accumulator is EXCLUSIVE ORed with the mask contained in the data memory location addressed by
register ‘r,’ bits 0-5 (8048} or bits 0-6 (8049).

Example:
XORDM: MOV  R1,#20H sMOVE 200 HEX TO REG 1

XRL A@R1 5 XOR’ ACC CONTENTS WITH MASK
;IN LOCATION 32

Accumulator{Immediate-Data Logical Operations

The mask to be combined logically with the accumulator value can be specified as ‘immediate’ data. This data is
recognized by the preceding pound sign (#) and must evaluate to eight bits. All instructions specifying immediate
data require two cycles for execution.

The result of the logical operation remains in the accumulator.

Logical AND Accumulator with Immediate Mask

Opcode Operands
ANL A #data
01 0 1|0 0 1 1 data

Data in the accumulator is logically ANDed with an immediately-specified mask.

Examples:

ANDID: ANL  A#0AFH ;' AND” ACC CONTENTS WITH
sMASK 10101111

ANL  A#3+X/Y ;' AND’ ACC CONTENTS WITH
sVALUE OF EXP 3 + X/Y’

Logical OR Accumulator With Immediate Mask

Opcode Operands
ORL A #data
01 0 0[O0 O 1 1 data

Data in the accumulator is logically ORed with an immediately-specified mask.

Example:

ORID: ORL A#X ;FOR” ACC CONTENTS WITH MASK
;01011000 (ASCIl VALUE OF ‘X’)

Logical XOR Accumulator With Immediate Mask

Opcode Operand's

XRL A #data

T 1 0 10 0 1 1 data
™

Data in the accumulator is EXCLUSIVE ORed with an immediately-specified mask.

Example:

XORID:  XOR A#HEXTEN ;XOR CONTENTS OF ACCWITH
;MASK EQUAL VALUE OF
;SYMBOL ‘HEXTEN’

Input/Output Logical Operations

Data residing on the BUS port or ports 1 and 2 can be logically combined with an immediately-specified mask. The
mask data must be preceded by ‘#’ and must evaluate to eight bits. Data on 8243 ports 4-7 can be logically com-
bined with a mask contained in bits 0-3 of the accumulator. In the case of the 8021 and 8022, 1/O logical
operations are permitted on 8243 ports only.

Only AND and OR logical operations can be done on i/O data. XOR is not possible. The results of the logical
operation remain on the specified port. Ali of the instructions described in this subsection require two cycles for
execution. These instructions can be used to clear/set any specified outputs.

Logical AND Port 1-2 With Immediate Mask

Opcode Operands
ANL Pp,#data p=1-2
T 0 0 1117 0 p p data

Data on port ‘p’ is logically ANDed with an immediately-specified mask. This instruction is not recognized by
the 8021 or 8022.

Example: -

ANDP2: ANL  P2,#0FOH ;AND’ PORT 2 CONTENTS WITH

;MASK ‘FO’ HEX (CLEAR P20-23) a
Logical AND BUS With Immediate Mask
Opcode Operand's
3 ANL BUS, #data
10 0 1|1 0 0 O data

Data on the BUS port is fogically ANDed with an immediately-specified mask. This instruction assumes prior speci-
fication of an ‘OUTL BUS,A’ instruction. The 8021 and 8022 do not recognize this instruction.

Example:

ANDBUS: ANL  BUS#MASK ;'AND’ BUS CONTENTS WITH
;MASK EQUAL VALUE OF
;SYMBOL ‘MASK?

Logical OR Port 1-2 With Immediate Mask

Opcode Operands
ORL Pp,#data p=1-2
1T 0 0 0j1 0 p p data

Data on port ‘p’ is logically ORed with an immediately-specified mask. This instruction is not recognized by
the 8021 or 8022.

Example:
ORP1: ORL P1,#0FFH ;OR” PORT 1 CONTENTS WITH
;MASK ‘FF’ HEX (SET PORT 1
;TO ALL ONES)
Logical OR BUS With Immediate Mask
Opcode Operand's
ORL BUS,#data
1T 0 0 Ol1T 0 0 O data

Data on the BUS port is logically ORed with an immediately-specified mask. This instruction assumes prior specifi-
cation of an ‘OUTL BUS,A’ instruction. The 8021 and 8022 do not recognize this instruction.

Example:
ORBUS: ORL BUS,#HEXMSK ;OR’ BUS CONTENTS WITH
;MASK EQUAL VALUE OF
;SYMBOL ‘HEXMSK?

Logical AND Port 4-7 With Accumulator Mask

Opcode Operand's

ANLD Pp,A p=4-7
[1 00 1|1 1 p p

Data on port ‘p’ is logically ANDed with the digit mask contained in accumulator bits 0-3 and the result written to
port ‘p.” The accumulator is not affected.

NOTE

The mapping of port ‘p’ to opcode bits 0-1 is as

follows:

7 0 Port

0 0 4

0 1 5

1 0 6

1 1 7

Example:
ANDP4: ANLD P4,A 'AND’ PORT 4 CONTENTS WITH

;ACC BITS 0-3

Logical OR Port 4-7 with Accumulator Mask
Opcode Operands

ORLD Pp,A p=4-7

10001 1 p p

Data on port ‘p’is logically ORed with the digit mask contained in accumulator bits 0-3 and the result is written to
port ‘p.” The accumulator is not affected. (See the NOTE accompanying the preceding instruction.)

Example:
ORP7: ORLD P7,A ;OR’ PORT 7 CONTENTS

;WITH ACC BITS 0-3

Rotate Operations

The MCS-48 instruction set includes four instructions for bit rotation of accumulator contents: right and left rota-

tions that do not affect the carry bit, and rotations through the carry. All four instructions perform ‘wraparound’
rotations, as shown in Figure 3-1.

Rotate Without Carry
Rotate Right Without Carry
Opcode

RR

Figure 3-1 Bit Rotation

Operand

The contents of the accumulator are rotated right one bit. Bit 0 is rotated into the bit 7 position (Figure 3-1).

Example: Assume accumulator contains 10110001.

RIGHT ROTATION,
CARRY UNAFFECTED

LEFT ROTATION,
CARRY UNAFFECTED

RIGHT ROTATION
THROUGH CARRY

LEFT ROTATION
THROUGH CARRY

RRNC: RR A ;NEW ACC CONTENTS ARE 11011000

Rotate Left Without Carry
Opcode

RL

Operand

Chapter 3. MCS-48 Assembly Language Instructions

The contents of the accumulator are rotated left one bit. Bit 7 is rotated into the bit 0 position (Figure 3-1).
Example: Assume accumulator contains 10110001.

RLNC: RL A ;NEW ACC CONTENTS ARE 01100011

Rotate Through Carry
Rotate Right Through Carry
Opcode Operand

RRC A

The contents of the accumulator are rotated right one bit. Bit 0 replaces the carry bit; the carry bit is rotated into
the bit 7 position (Figure 3-1).

Example: Assume carry is not set and accumulator contains 10110001.

RRTC: RRC A ;CARRY ISSET AND ACC
;CONTAINS 01011000

Rotate Left Through Carry

Opcode Operand

RLC A

The contents of the accumulator are rotated left one bit. Bit 7 replaces the carry bit; the carry bit is rotated into the
bit 0 position (Figure 3-1).

Example: Assume accumulator contains a ‘signed’ number; isolate sign without changing value.

RLTC: CLR C ;CLEAR CARRY TO ZERO
RLC A ;ROTATE ACC LEFT;SIGN
;BIT (7) IS PLACED IN CARRY *
RR A ;ROTATE ACC RIGHT — VALUE
i(BITS 0-6) IS RESTORED, CARRY
;UNCHANGED, BIT 7 IS ZERO

Arithmetic Operations

Arithmetic operations include the increment, decrement, and addition instructions.

Increment/Decrement Instructions

You can increment (by one) the contents of the accumulator, a working register, or resident data memory location.
The accumulator and working registers can be decremented. (External data memory contents cannot be incremented

or decremented directly, although such data can be manipulated in the accumulator.)

The DJNZ instruction allows you to decrement a register, test for zero, and transfer program control accordingly.

The register can be used as a counter, providing program loop control.

Increment Accumulator
Opcode Operand

INC A

000 1T)0 1T 11

The contents of the accumulator are incremented by one.

Example: Increment contents of location 100 in external data memory.

INCA: MOV  RO,#100 ;MOVE 100’ DEC TO
;ADDRESS REG 0
MOVX A,@RO ;MOVE CONTENTS OF LOCATION
;100 TO ACC
INC A ;INCREMENT A
MOVX @RO,A ;MOVE ACC CONTENTS TO

;LOCATION 100

Increment Register
Opcode Operand

INC Rr r=0-7

O 0 0 11T r r r

The contents of working register ‘r’ are incremented by one.
Example:

INCRO: INC RO ;INCREMENT ADDRESS REG 0

Increment Data Memory Location

Opcode Operand
INC @Rr r=1-2
000 1|0 0 0 r -

The contents of the resident data memory location addressed by register ‘r’ bits 0-5 (8048) or bits 0-6 (8049) are
incremented by one.

Example:

INCDM: MOV  R1,#3FH ;MOVE ONES TO BITS 0-5
INC  @R1 ;INCREMENT LOCATION 63

Decrement Accumulator

O;;code Operand

DEC A

0 00 00 1T 11

The contents of the accumulator are decremented by one.

Example: Decrement contents of external data memory location 63.

MOV RO,#3FH ;MOVE 3F"HEX TO REG 0
MOVX A,@RO ;MOVE CONTENTS OF LOCATION
;63 TO ACC
DEC A ;DECREMENT ACC
MOVX @RO,A ;MOVE CONTENTS OF ACC TO LOCATION

;63 IN EXPANDED MEMORY

Decrement Register

Opcode Operand

DEC Rr r=0-7

The contents of working register ‘r’ are decremented by one. This instruction is not recognized by the 8021,

Example:

DECR1: DEC RI1 ;DECREMENT ADDRESS REG 1

3-24



Chapter 3. MCS-48 Assembly Language Instructions

Decrement Register and Test

Opcode Operand
DJNZ Rr,address r=0-7
11T 01t roror address

This isa 2-cycle instruction. Register ‘r’ is decremented and tested for zero. If the register contains all zeros, program
control falls through to the next instruction. If the register contents are not zero, control jumps to the specified
‘address.’

The address in this case must evaluate to eight bits, that is, the jump must be to a location within the current 256-
location page.

NOTE

A 12-bit address specification does not cause an error
if the DJNZ instruction and the jump target are on the
same page. If the DJNZ instruction begins in location
255 of a page, it must jump to a target address on the
following page.

Example: Increment values in data memory locations 50-54.

MOV RO,#50 ;MOVE ‘50’ DEC TO ADDRESS REG 0
MOV R3,#5 ;MOVE 5’ DEC TO COUNTER REG 3
INCRT: INC @RO SINCREMENT CONTENTS OF LOCATION
;ADDRESSED BY REG 0
INC RO ;INCREMENT ADDRESS IN REG 0

DINZ R3,INCRT ;DECREMENT REG 3 — JUMP TO
“INCRT' IF REG 3 NONZERO

NEXT - “NEXT’ ROUTINE EXECUTED IF
:R3 IS ZERO

Addition Instructions

The contents of working registers or other resident data-memory locations, or immediately-specified data, can be
added to the contents of the accumulator. The result remains in the accumulator.

As described earlier, data memory locations are addressed indirectly through registers 0-1. The reference to these regis-
ters is preceded by ‘@’ to indicate indirection. Immediately-specified data is preceded by ‘#’ and must evaluate to
eight bits. All immediate operations require two cycles for execution.

Addition can be performed ‘with carry.” This means that the value in the carry bit is added to the accumulator at the
low-order end and the carry bit is set to zero automatically before the regular addition operation takes place. This is
necessary, for example, when adding 16-bit values, to ensure that any carry from the low-order byte addition is re-
flected in the high-order byte addition.

Example: Add value 10101010 to accumulator value 10000010 with carry. Assume carry bit is currently set.
STEP1: ADDCto ACC and zero C

C 7 ACC 0

@ [To10[1 011

STEP 2:  ADD 10000010 to ACC; overflow to C if necessary

C 7 ACC 0

floo1o[110T7]

All addition operations (with or without carry) affect the carry and auxiliary carry bits in the event of an addition
overflow.

Add Register Contents to Accumulator
Opcode Operands

ADD ARr r=0-7

The contents of register r’ are added to the accumulator.

Example:

ADDREG: ADD AR6 ;ADD REG 6 CONTENTS TO ACC

Add Carry and Register Contents to Accumulator
Opcode Operands
ADDC A Rr r=0-7

01 1 1[1 rr r]

The content of the carry bit is added to accumulator bit O and the carry bit cleared. The contents of register ‘r’ are
then added to the accumulator.

Example:

ADDRGC: ADDC AR4 ;ADD CARRY AND REG 4 CONTENTS
;TO ACC

Add Data Memory Contents to Accumulator
Opcode Operand's

ADD A@Rr r=0-1

01 1 0|0 0 0 r

The contents of the standard data memory location addressed by register ‘r’ bits 0-5 (8048) or bits 0-6 (8049) are
added to the accumulator.

Example:
ADDM: MOV RO,#2FH sMOVE 2F’ HEX TO REG 0
ADD A,@R0 ;ADD VALUE OF LOCATION 47 TO
;ACC

Add Carry and Data Memory Contents to Accumulator
Opcode Operands

ADDC A,@Rr r=0-1

O 1 1T 110 0 0 r

The content of the carry bit is added to accumulator bit O and the carry bit is cieared. Then the contents of the

standard data memory location addressed by register ‘r’ bits 0-5 (8048) or bits 0-6 {8049) are added to the
accumulator.

Example:
ADDMC: MOV R1,#40 ;MOVE ‘40' DEC TO REG 1
ADDC A,@RI1 ;ADD CARRY AND LOCATION 40
;CONTENTS TO ACC

Add Immediate Data to Accumulator

Opcode Operands
ADD A Fdata
0 00 0|0 O 1 1 data

This is a 2-cycle instruction. The specified data is added to the accumulator.

Example:

ADDID: ADD A#ADDER ;ADD VALUE OF SYMBOL
;ADDER’ TO ACC

Add Carry and Immediate Data to Accumulator

Opcode Operands
ADDC A #data
0 00 1T]0 0 1 1 data

This is a 2-cycle instruction. The content of the carry bit is added to accumulator location O and the carry bit
cleared. Then the specified data is added to the accumulator.

Example:

ADDIDC: ADDC A,#225 ;ADD CARRY AND 225’
;DEC TO ACC

Miscellaneous Accumulator Operations

Three data manipulation instructions allow the accumulator contents to be cleared, complemented, or divided into
two decimal digits.

Clear Accumulator

Opcode Operand

CLR A

The contents of the accumulator are cleared to zero.

Complement Accumulator

Opcode Operand

CPL A

«*
The contents of the accumulator are complemented. This is strictly a one’s complement. Each one is changed to zero
and vice-versa. {See the discussion of arithmetic notation in Chapter 2, the subsection ‘Two’s Complement Arithmetic.’)
Example: Assume accumulator contains 01101010.
CPLA: CPL A ;ACC CONTENTS ARE .

;COMPLEMENTED TO 10010101

Decimal Adjust Accumulator

Opcode Operand

DA A

" o1 0 1]o 1 1 1]

The 8-bit accumulator value is adjusted to form two 4-bit Binary Coded Decimal (BCD) digits (basically following an
addition operation). The carry bit is affected.

If the contents of bits 0-3 are greater than nine, or if the auxiliary carry bit is one, the accumulator is incremented
by six.

‘ The four high-order bits are then checked. If bits 4-7 exceed nine, or if C is one, these bits are increased by six. If an
overflow occurs, C is set to one; otherwise, it is cleared to zero.

Example:  Assume accumulator contains 10011011.
DA A ;ACC ADJUSTED TO 00000001 WITH C SET

C AC 7 4

O o[ oo |
o o7 |

0110 ADD SIX TO BITS 4-7
[1]o] [0 0oo0o0] 0001 OVERFLOWTOC

ADD SIX TO BITS 0-5

SETTING PROGRAM CONTROLS

section ‘Programmable Controls.” This section describes the instructions for manipulating these controls. it also des-
cribes interrupt controls, timer/event-counter controls, clock control, the selection of memory and register banks,
and the NOP instruction.

‘ Your program can be controlled by the setting of the condition bits, flags, and switches described in Chapter 1, the

Carry and Flag Controls

The carry bit (C), flag 0 (FO), and flag 1 (F1) can be cleared or complemented by the following instructions. Carry
(PSW bit 7) and flag O (PSW bit 5) can also be manipulated by moving the PSW to the accumulator, masking the
entire eight bits, then moving the result back to the PSW. This might be a preferable approach if several other bits in
the PSW were being altered at the same time.

Clear Carry Bit
‘ Opcode Operand
CLR C

[t 0o 0o 1]o 1 1 1]

During normal program execution, the carry bit can be set to one by the ADD, ADDC, RLC, RRC, CPL C, and DA

instructions. This instruction resets the carry bit to zero.

Complement Carry Bit

Opcode Operand

CPL C

[1 0o 1 ofo 1 1 1

The setting of the carry bit is complemented; one is changed to zero, and zero is changed to one.

Example: Set C to one; current setting is unknown.

;C IS CLEARED TO ZERO

CT01: CLR C
;C IS SET TO ONE

cPL C

Clear Flag 0

Opcode Operand

CLR FO

10 0 0[O0 1 0 1

Flag O is cleared to zero. The 8021 and 8022 do not recognize this instruction.

Complement Flag 0

Opcode Operand

CPL FO

[1 00 1]o 1 01

The setting of flag 0 is complemented; one is changed to zero, and zero is changed to one. The 8021 and 8022

do not recognize this instruction.

Clear Flag 1

Opcode Operand

CLR F1

1.0 1 0[0 1 0 1]

Flag 1 is cleared to zero. The 8021 does not recognize this instruction.

. Complement Flag 1

Opcode Operand
CPL F1
1T 0 1 1 1 1
» 0 0
| The setting of flag 1 is complemented; one is changed to zero, and zero is changed to one. The 8021 and 8022
do not recognize this instruction.
Interrupt Controls
As described in Chapter 1, the 8048 responds to two kinds of interrupts: an ‘external’ interrupt initiated by a low
. signal on the interrupt input pin, and an overflow in the timer/event-counter register. The following instructions

allow you to enable and disable these interrupts.
These interrupts and related instructions are not available on the 8021.

External Interrupt Control

If the external interrupt is enabled and the interrupt input pin goes to level zero, the interrupt sequence is activated.
Control passes to program memory location 3, the program counter and bits 4-7 of the PSW are stored in the program
stack, and the stack pointer (PSW bits 0-2) is incremented by one. Since there is no PSW on the 8022, the

. accumulator and carry flags can be saved using software as described in chapter 1 under “8022 Microcomputer
Software Differences.”

Enable External Interrupt

Opcode Operand

. EN I
[0000J010j

External interrupts are enabled. A low signal on the interrupt input pin initiates the interrupt sequence. This
instruction is not recognized by the 8021.

Disable External Interrupt

Opcode Operand
E
DIS l
o 0 0o 1]o 1 0 1]
External interrupts are disabled. A low signal on the interrupt input pin has no effect. This instruction is not
recognized by the 8021.

Timer[Counter Interrupt Control
If this interrupt is enabled and the timer/event-counter overflows, the interrupt sequence is activated. Control passes
to program memory location 7, the program counter and PSW bits 4-7 are stored in the program stack, and the stack

pointer incremented.
The timer flag (TF) is set when the timer/counter overflows, whether or not the interrupt is enabled. The timer
continues to accumulate time after an overflow occurs.
Enable Timer[Counter Interrupt
Opcode Operand

EN TCNTI

oo 1 0{0 1 01

Timer/counter interrupts are enabled. An overflow of this register initiates the interrupt sequence. This
instruction is not recognized by the 8021.

Disable Timer/Counter Interrupt

Opcode Operand

DIS TCNTI

001 1/01 01

Timer/counter interrupts are disabled. Any pending timer interrupt request is cleared. The interrupt sequence is not
initiated by an overflow, but the timer flag is set and time accumulation continues. This instruction is not
recognized by the 8021.

Timer/Event-Counter Controls

The following instructions are used to start and stop time accumulation or event counting in the timer/event-counter
register.

Start Timer
Opcode Operand ~

STRT T

lo 1 0 1[0 1 0 1

Timer accumulation is initiated in the timer register. The register is incremented every 32 instruction cycles. The pre- .
scaler (where the 32 cycles are counted) is cleared, but the timer register is not.

Example: Initialize and start timer.

STARTT: CLR A

MOV T,A
EN TCNTI
STRT T

Start Event Counter

Opcode

STRT

Chapter 3. MCS-48 Assembly Language Instructions

;CLEAR ACC TO ZEROS
;MOVE ZEROS TO TIMER
;ENABLE TIMER INTERRUPT
;START TIMER

Operand

CNT

0100{0101]

The test 1 (T1) pin is enabled as the event-counter input and the counter is started. The event-counter register is
incremented with each high-to-low transition on the T1 pin. For the 8022 only, the eventcounter register is
incremented with each low-to-high transition on T1.

Example: Initialize and start event counter. Assume overflow is desired with first T1 input.

STARTC: MOV  A#0FFH sMOVE ‘FF’ HEX (ONES) TO

;ACC
MoV  TA ;MOVE ONES TO COUNTER
EN TCNTI ;ENABLE COUNTER INTERRUPT
STRT CNT ;ENABLE T1 AS COUNTER

;INPUT AND START

Stop Timer/Event Counter
Opcode Operand
STOP TCNT

0 1 1 0jo 1 01

This instruction is used to stop both time accumulation and event counting.

Example: Disable interrupt, but jump to interrupt routine after eight overflows and stop timer. Count overflows in .
register 7.

START: DIS TCNTI ;DISABLE TIMER INTERRUPT
CLR A ;CLEAR ACC TO ZEROS
MOV  T,A ;MOVE ZEROS TO TIMER
MOV  R7,A ;MOVE ZEROS TO REG 7

MAIN:  STRT T START TIMER -
JTF COUNT ;JUMP TO ROUTINE ‘COUNT’

;iIF TF=1 AND CLEAR TIMER FLAG

JMP MAIN ;CLOSE LOOP

COUNT: INC  R7 ;INCREMENT REG 7
MOV  A,R7 :MOVE REG 7 CONTENTS TO ACC
JB3 INT ;JMUP TO ROUTINE ‘INT’ IF
;ACC BIT 3 1S SET (REG 7=8)
JMP  MAIN OTHERWISE RETURN TO ROUTINE
MAIN’
INT: STOP  TCNT ;STOP TIMER
JMP 7H :JUMP TO LOCATION 7

;(TIMER INTERRUPT ROUTINE)

Clock Control

The test 0 (TO) pin can be used as a state clock output and tested directly by your program. See the MCS-48
user’s manual for details. Neither the 8021 nor the 8022 support this feature.

Enable Clock Output

Opcode Operand

ENTO CLK -

011 1[0 1 01

The test O pin is enabled to act as the clock output. This function is disabled by a system reset. The 8021 and
8022 do not recognize this instruction.

Example:

ENTSTO: ENTO CLK ;ENABLE T0 AS CLOCK OUTPUT

Memory and Register Bank Controls
The following instructions allow you to control the interpretation of program memory references and references to
data memory working registers. As noted in Chapter 1, memory and register bank selection is not possible on the

8021 or 8022. It always refers to bank ‘0’

Memory Bank Selection

The memory bank instructions let you specify your program memory address references to be in ‘bank 0’ (locations
0-2047) or ‘bank 1’ (locations 2048-4095). See Figure 1-1. These instructions toggle program counter bit 11, but not
until the next branch from the main program (via a jump or call) begins execution.

11 PROGRAM COUNTER 0

SelectBank0 [0 0 1 0[1 0 0 0[1 0 1 0]

Select Bank 1 [1010|10001010]

Location 650

Location 2698

i

if a SEL MB instruction is issued before a CALL, it affects only the subroutine called. The return restores PC bit 11
to its previous value (see NOTE 1). A SEL MB issued before a jump instruction modifies PC bit 11 permanently.

NOTES

1. While PC bit 11 is restored on returning from a CALL, the
‘designate bank’ internal flip-flop (DBF) is not. This means
you mustreset the DBF with another SEL before issuing
another jump instruction.

2. When an interrupt service routine is executing, program
counter bit 11 is held at zero. This means any service
routine references must reside in memory bank 0. The
select-memory-bank instructions should not be issued in

an interrupt service routine.

The initial value of PC bit 11 is zero and merhory bank 0 is selected.

Select Memory Bank 0
Opcode Operand
SEL MBO

[T 1 1 0]0o 1 0 1

PC bit 11 is set to zero. All references to program memory addresses fall within the range 0-2047. This instruction
is not recognized by the 8021 or the 8022.

Example: Assume program counter contains 834H (2100D).

SEL MBO ;SELECT MEMORY BANK O
JMP  $+20 :)JUMP TO LOCATION 48H (72D)
Select Memory Bank 1 -
Opcode Operand
SEL MB1 ’

(111ﬂ0101

PC bit 11 is set to one. All references to program memory addresses fall within the range 2048-4095. This
instruction is not recognized by the 8021 or the 8022.

Register Bank Selection

The register bank instructions let you specify whether references to registers 0-7 address data memory locations in
‘vank 0’ (locations 0-7) or ‘bank 1’ (locations 24-31). See Figure 1-2. These instructions toggle the register bank
switch (PSW bit 4). The initial setting of this bit is zero.

Select Register Bank 0

Opcode Operand

SEL RBO

T 1.0 0,0 1 0 1

PSW bit 4 is set to zero. References to working registers 0-7 address data memory locations 0-7. This is the recom-
mended setting for normal program execution. The 8021 and 8022 do not recognize this instruction.

Select Register Bank 1

Opcode Operand -
SEL RB1
-
110 1Jo 1 0 1
PSW bit 4 is set to one. References to working registers 0-7 address data memory locations 24-31. This is the recom-
mended setting for interrupt service routines, since locations 0-7 are left intact. The RETR instruction at the end of
the interrupt service routine restores bit 4 of the PSW to the value it had at the time of the interrupt. .

The 8021 and 8022 do not recognize this instruction.

Example: Assume an external interrupt has occurred, control has passed to program memory location 3, and PSW
bit 4 (register bank switch) was zero before the interrupt.

LOC3: JMP INIT ;JUMP TO ROUTINE “INIT’ [F
;INTERRUPT HAS OCCURRED

INIT: Mov R7,A ;MOVE ACC CONTENTS TO

;LOCATION 7
SEL RB1 ;SELECT REG BANK 1
MOV R7,#0FAH ;MOVE ‘FA’ HEX TO LOCATION 31
SEL RBO ;SELECT REG BANK 0
MOV A,R7 ;RESTORE ACC FROM LOCATION 7
RETR ;RETURN — RESTORE PC AND PSW
47

The ‘Null’ Operation
The null operation uses one machine cycle, but no operation is performed. Its primary function is to reserve a pro-

gram location for an instruction to be inserted later. It could aiso be used, like the timer, to synchronize your
system.

The NOP Instruction
Opcode

NOP

oooojooog

No operation is performed. Execution continues with the following instruction.

TRANSFERRING PROGRAM CONTROL

Instructions in program memory are normally executed sequentially. Program control can be transferred out of the
main line of code by an external or timer interrupt, or when a jump or call instruction is encountered.

An interrupt (if enabled) automatically transfers control to location 3 (for external interrupts) or location 7 {for

timer overflows), and is essentially a call to an interrupt service subroutine. The program counter and PSW bits 4-7
are saved in the stack.

Your program can save the accumulator and flags in software for the 8021/8022 as described in Chapter 1
under “8022 Microcomputer Software Differences.”

Your program can also contain other subroutines to perform frequently-executed code. Control is passed to these
subroutines by the CALL instruction, which also saves the program counter and PSW bits 4-7.

Control is returned from an interrupt service routine or other subroutine to the main program by the RET and
RETR instructions. RET restores only the program counter; RETR restores both the program counter and PSW
bits 4-7.

Subroutines are entered and exited using the CALL, RET, and RETR instructions. For the 8022 only, the RET]
instruction must be used to return from an interrupt.

The jump instructions alter the contents of the program counter without saving PC or PSW information. jumps can
be specified subject to certain conditions (such as the setting of a flag), or can be made unconditional.

All conditional jumps and the JMPP instruction limit the range of a jump to the current 256-location page (that is,
aiter PC bits 0-7). The JMP and CALL instructions allow program control to be transferred within a 2K memory
bank (that is, alter PC bits 0-10). This range can be extended to 4K by toggling PC bit 11 with the SEL MB instruc-
tions. A SEL MB preceding a CALL instruction is valid only for the duration of the subroutine; a SEL MB preceding
a jump remains in effect until changed by your program.

Jump instructions with 8-bit operands imply a destination address expressable in 12 bits. All 8-bit addresses are valid;
12-bit destination addresses are valid if the jump instruction and destination reside on the same page. If a conditional
jump or JMPP begins in focation 255 of a page, it must reference a destination on the following page. Any jump
instruction beginning in location 2047 or 4095 is invalid. A CALL cannot begin in locations 2046-2047 or 4094-4095.

All control transfer and return instructions require two cycles for execution.

Subroutine Call/Return Operations

Subroutines are entered and exited using the CALL, RET, and RETR instructions.
For the 8022 only, RETI is used to return from an interrupt, and RET is used for all other returns.

Subroutine Call

Opcode Operand

CALL address
10 8 7 0
| addr [1 010 0 addr |

The program counter and PSW bits 4-7 are saved in the stack. The stack pointer (PSW bits 0-2) is updated. Program
control is then passed to the location specified by ‘address.’ PC bit 11 is determined by the most recent SEL MB
instruction. PC bits 10-11 must always be ‘0’ for the 8021 or a ‘range’ error (R} results.

Execution continues at the instruction following the CALL upon return from the subroutine.

Example: Add three groups of two numbers. Put subtotals in locations 50, 51 and total in location 52.

MOV  RO,#50 ;MOVE ‘50’ DEC TO ADDRESS

;REG 0
BEGADD: MOV  AR1 ;MOVE CONTENTS OF REG 1 TO

;ACC

ADD AR2 ;ADD REG 2 TO ACC

CALL SUBTOT ;CALL SUBROUTINE ‘SUBTOT’

ADD  AR3 ;ADD REG 3 TO ACC

ADD AR4 ;ADD REG 4 TO ACC

CALL SUBTOT ;CALL SUBROUTINE ‘SUBTOT’

ADD AR5 ;ADD REG 5 TO ACC

ADD  AR6 ;ADD REG 6 TO ACC
CALL SUBTOT ;CALL SUBROUTINE ‘SUBTOT’

SUBTOT: MOV  @RO,A ;MOVE CONTENTS OF ACC TO
;LOCATION ADDRESSED BY REG 0

INC RO ;INCREMENT REG 0
RET ;RETURN TO MAIN PROGRAM

Return Without PSW Restore
Opcode

RET

|1ooooo1ﬂ

The stack pointer (PSW bits 0-2) is decremented. The program counter is then restored from the stack. PSW bits 4-7
are not restored.

Return With PSW Restore
Opcode

RETR

T 0 0 170 0 1 1

The stack pointer is decremented. The program counter and bits 4-7 of the PSW are then restored from the stack.
Note that RETR should be used to return from an interrupt, but should not be used within the interrupt service
routine as it would clear the interrupt in process.

This instruction is not recognized by the 8021 or the 8022.

Return from Interrupt (8022 only)

Opcode
RETI

1 00 1[0 0 1 1

This is a 1-byte, 2-cycle instruction. It is recognized only by the 8022 and should be used in 8022 programs only
when returning from an interrupt. All other returns on the 8022 should be performed using the RET instruction.

Jump Instructions

The MCS-48 instruction set includes two unconditional jumps and 13 conditional jumps (in addition to the DJNZ

instruction described earlier in this chapter). Only one jump instruction, JMP, alters PC bits 0-10. The others affect
only PC bits 0-7 and, therefore, must address a location within the current 256-location page.

Unconditional Jumps

The JMP unconditional jump allows you to cross page boundaries; JMPP is limited to the current page. JMP
addresses program memory locations directly; JMPP addresses program memory locations indirectly through
the accumulator. Indirection is indicated by prefixing the accumulator reference with a ‘commercial at’ (@).

Direct Jump Within 2K Block

Opcode Operand
JMP address
10 8 7 0
[addrf0 01 00 addr ]

Bits 0-10 of the program counter are replaced with the directly-specified address. The setting of PC bit 11 is
determined by the most recent SELECT MB instruction. PC bits 10-11 must always be ‘O’ for the 8021 or a
‘range’ error (R) results.

Examples:
JMP  SUBTOT ;JUMP TO SUBROUTINE ‘SUBTOT’
JMP  $-6 ;JUMP TO INSTRUCTION SIX LOCATIONS
;BEFORE CURRENT LOCATION
JMP  2FH ;JUMP TO ADDRESS 2F’ HEX

Indirect Jump Within Page

Opcode Operand
JMPP @A

[1011|001ﬂ

The contents of the program memory location pointed to by the accumulator are substituted for the ‘page’ portion
of the program counter (PC bits 0-7).

Example: Assume accumulator contains OFH.

JMPPAG: JMPP @A ;JUMP TO ADDRESS STORED IN
;LOCATION 15 IN CURRENT PAGE

Conditional Jumps

The following jumps are executed only if a specific condition is satisfied. All jumps occur within the current page.

Jump If Carry Is Set

Opcode Operand
JjC address
T 1 1 110 1 1 0 address

Control passes to the specified address if the carry bit is set to one.
Example:

JC1: JC OVFLOW ;JUMP TO ‘OVFLOW’ ROUTINE IF C=1

Jump If Carry Is Not Set
Opcode Operand
JNC address

[1 1 1ﬂo 11 ﬂ address

Control pzisses to the specified address if the carry bit is not set, that is, equals zero.

Example:

JCO: JNC NOVFLO ;JUMP TO ‘NOVFLO’ ROUTINE IF C=0

Jump If Accumulator Is Zero

Opcode Operand
1z address
1 1 00[01 10 address |

Control passes to the specified address if the accumulator contains all zeros when this instruction is executed. Accumu-
lator contents are monitored continuously.

Example:

JACCO: |Z OA3H ;JUMP TO LOCATION ‘A3’ HEX
;IF ACC VALUE IS ZERO

342

Jump If Accumulator Is Not Zero

Opcode Operand
JNZ address
1 00 1|0 1T 1 O address

Control passes to the specified address if the accumulator contents are nonzero when this instruction is executed.
Accumulator contents are monitored continuously.

Example:
JACCNO: JNZ OABH ;JUMP TO LOCATION ‘AB’ HEX

;IF ACC VALUE 1S NONZERO

Jump If Flag O Is Set

Opcode Operand
JFO address
1T 01 110 1 1 0 address

Control passes to the specified address if flag O is set to one. This instruction is not recognized by the 8021 or
the 8022.

Example:
JEOIST:  JFO TOTAL ;JUMP TO ‘TOTAL' ROUTINE IF
;FO=1
Jump If Flag 1 Is Set
Opcode Operand
JF1 address
o1 1 10 1T 1 0 address

Control passes to the specified address if flag 1 is set to one. This instruction is not recognized by the 8021 or
the 8022.

Example:

JF11S1:  JF1  FILBUF ;JUMP TO ‘FILBUF' ROUTINE
SIF F1=1

Jump If Test O Is High

Opcode Operand
JTO address
0O 01 170 1 1 0 address

Control passes to the specified address if the test 0 signal is high (=1). This instruction is not recognized by
the 8021.

Example.
JTOHI: JTO 53 ;JUMP TO LOCATION 53 DEC IF
;T0=1
Jump If Test 0 Is Low
Opcode Operand
JNTO address
0 01T 00 1T 1 0 address

Control passes to the specified address if the test 0 signal is fow (=0). This instruction is not recognized by
the 8021.

Example:
JTOLOW: JNTO 60 ;JUMP TO LOCATION 60
;DEC IF T0=0
Jump If Test 1 Is High
Opcode Operand
JTh address
01 0 110 1T 1 0 address

Control passes to the specified address if the test 1 signal is high (=1).
Example:

JTHHL: JTT COUNT ;JUMP TO 'COUNT' ROUTINE
SIFT1=1

Jump If Test 1 Is Low

Opcode Operand
JNT1 address
01 00[01 10 address

Control passes to the specified address if test 1 signal is low (=0).

Example:
JTILOW: JNT1 NOCNT ;JUMP TO ‘NOCNT’ ROUTINE
;IF T1=0
Jump If Timer Flag Is Set
Opcode Operand
JTF address
o oo 1]o 1 1 o address |

Control passes to the specified address if the timer flag is set to one, that is, the timer/counter register has over-
flowed. Testing the timer flag resets it to zero. (This overflow initiates an interrupt service sequence if the timer-
overflow interrupt is enabled.)

Example:
JTF1: JTF TIMER ;JUMP TO ‘TIMER’ ROUTINE
IFTF=1
Jump If Interrupt Input Is Low
Opcode Operand
JNI address

1 0 0 0[0 1 1 o]  address ]

Control passes to the specified address if the interrupt input signal is low (=0}, that is, an external interrupt has been
signalled. (This signal initiates an interrupt service sequence if the external interrupt is enabled.) The 8021
and 8022 do not recognize this instruction.

Example: The JNI instruction is used to control a test input.

DIS I ;DISABLE EXTERNAL INTERRUPT
JNI TRUE ;JUMP TO “TRUE’ ROUTINE

JIF 1=0 (TEST IS TRUE)
JMP $—2 ;LOOP TO NI TEST

Jump If Accumulator Bit Is Set

Opcode Operand
JBb address b=0-7
(b b b 1|0 0 1 0 address

Control passes to the specified address if accumulator bit ‘b’ is set to one. The 8021 and 8022 do not recognize
this instruction.

Example:

JB4IS1:  |B4 NEXT ;JUMP TO 'NEXT’ ROUTINE
JIF ACC BIT 4=1

SAMPLE PROGRAMS

The following examples demonstrate addition, subtraction, multiplication, and number comparison using 8-bit,
16-bit, and BCD quantities. Analog/digital conversion and a keyboard scan are also demonstrated.

Addition With 8-Bit Quantities

Add 8-bit symbolic values ADDEND and AUGEND and place their sum in Register 7.

ADDS: MOV  A,#ADDEND
ADD  A#AUGEND
MOV R7,A

Addition With 16-Bit Quantities

Add two 16-bit numbers and store their sum in registers 6 (high-order byte) and 7 (low-order byte).

ADD16: MOV A #ADDLOW
ADD  A#AUGLOW
MOV  R7,A

MOV A #ADDHI

ADDC A,#AUGH! ;INCLUDE OVERFLOW FROM
;PREVIOUS ADD IN ADDITION

MOV  R6,A

Addition With BCD Quantities
Add the BCD number whose LSD is at focation BETA to the BCD number whose LSD is at location ALPHA and

store the result in ALPHA. Length of number is ‘COUNT’ digit pairs. For this example, assume both numbers are
the same length and have an even number of digits (or the most-significant digit is zero, if odd).

ADDBCD: MOV  RO,#ALPHA ;AUGEND, SUM LSD
;LOCATION IN REG 0
MOV  R1,#BETA ;ADDEND LOCATION
;INREG 1
MOV  R2,#COUNT ;LOOP COUNTER IN
;REG 2
CLR C
LOOP: MOV  A,@RO ;ADD ROUTINE -
ADDC A,@R1
DA A
MOV  @RO,A ;STORE SUM -
DEC RO ;DECREMENT ADDRESS
;REGS
DEC RI
DJNZ R2,LOOP ;LOOP CONTROL

Subtraction With 8-Bit Quantities

Subtract 8-bit subtrahend from 8-bit minuend using two’s complement addition and store difference in register 7.

SUBS: MOV  A#SUBHND
CPL A ;ONE’S COMPLEMENT A
INC A TWO’S COMPLEMENT A
ADD A #MINEND
MOV  R7,A

Subtraction With 16-Bit Quantities

Subtract two 16-bit numbers and store their difference in registers 3 (high-order byte) and 4 (low-order byte). Note
the use of ADD, rather than INC, to form the two’s complement numbers; INC does not affect the carry bit.

SUB16: MOV A #SUBLOW .
CPL A
ADD  A# ;FORM TWO’S COMPLEMENT
MOV  R4,A ;STORE TEMP SUBLOW COMP
MOV A #SUBHI
CPL A .
ADDC A, #0 ;PICK UP OVERFLOW AND
;FORM TWO’ S COMPLEMENT
MOV  R3,A ;STORE TEMP SUBHI COMP -
MOV  AR4 ;BEGIN ADDITION
ADD A #MINLOW
MOV  R4,A ;STORE LOW-ORDER DIFF
MOV  AR3
ADDC A #MINHI

MOV  R3,A ;STORE HIGH-ORDER DIFF

. Multiplication (8 X 8 Bits, 16-Bit Product)

Multiply two 8-bit numbers and store the 16-bit product in registers 2 and 3. Note than nine shifts through the
accumulator are required to justify the product correctly.

MPY8X8: MOV  R5,#9 ;8 + 1IN LOOP COUNTER
MOV  R6,#MCAND sMULTIPLICAND IN REG 6
@€ MOV  R3,#MPLIER sMULTIPLIER, LOW PARTIAL
;PRODUCT IN REG 3
CLR A
s CLR C
LOOP: RRC A ;ROTATE
XCH AR3 ; CARRY, ACC, REG 3
RRC A ; RIGHT
. XCH  A,R3 ; ONE BIT
JNC NOADD ;TEST CARRY
ADD AR6
NOADD: DJNZ R5,LOOP ;9 SHIFTS TO JUSTIFY
MOV  R2A ;STORE HIGH PARTIAL
;PRODUCT
Compare Memory to Accumulator
. Make an unsigned comparison between the contents of a memory location and the accumulator. Save original

accumulator contents temporarily in register 5.

COMPAR: MOV  R5A

MOV  RO,#MEM ;ADDRESS OF NUMBER TO BE
;,COMPARED

CPL A

INC A

ADD  A,@RO ;ACC CONTENTS DESTROYED
. 1z EQUAL ACC = MEM

JNC  ACCGT ;ACC GREATER THAN MEM

JC ACCLT ;ACC LESS THAN MEM

Analog/Digital Conversion

Construct an A/D converter from a D/A converter, a comparator op-amp, and a successive-approximation software
> | routine. A/D conversion (on-chip) for the 8022 is described in Chapter 1 under “8022 Microcomputer.”

The 8048 sends eight bits of data to the D/A converter via output port 1. The output of the D/A converter is com-
pared to the ‘analog input’ being converted. The result of the comparison (0 if DAC output is lower, 1 if higher) is
sent back to the 8048 via the TO input pin for handling. This procedure lets the 8048 estimate the proper digital
representation of the analog input by testing the most significant bit, keeping it if the input says ‘still too low’ or

. dropping it if the input says ‘too high now.’ From this point, each bit is tested in order of significance and either
kept or discarded.

ANALOG INPUT

Vay ¥ VIN

OP-AMP QUTPUT IS B IF V5 < VN
1IF V2 > VIN

P1 / > DAC
/ 8
8048
1 TO

